
 Abstract

This paper is a mini-user guide for DFSORT's versatile ICETOOL data processing and reporting utility. The major
features of ICETOOL for z/OS DFSORT V1R10 (used for z/OS 1.10 and z/OS 1.11) and z/OS DFSORT V1R12
(used for z/OS 1.12), including its JCL and control statements, are discussed at length using many examples. The
objective is to show you how to use DFSORT's ICETOOL to accomplish complex tasks.

ICETOOL Mini-User Guide

 Introduction

ICETOOL, a versatile data set processing and reporting utility, provides an easy-to-use batch front-end for
DFSORT. ICETOOL combines new features with previously available DFSORT features to perform complex
sorting, copying, merging, reporting and analytical tasks using multiple data sets in a single job step. ICETOOL
was first introduced in DFSORT Release 11.1 and was significantly enhanced in each subsequent DFSORT release.
The many enhancements to ICETOOL through October, 2010 are reflected in this paper.

This paper is a mini-user guide for ICETOOL. The major features of ICETOOL for z/OS DFSORT V1R10 (used
for z/OS 1.10 and z/OS 1.11) and z/OS DFSORT V1R12 (used for z/OS 1.12), including its JCL and control
statements, are discussed at length using many examples. The objective is to show you how to use DFSORT's
ICETOOL to accomplish complex tasks.

Complete information on ICETOOL's JCL, control statements, restrictions, and calling program interface can be
found in DFSORT Application Programming Guide (SC26-7523). "Examples of DFSORT Job Streams" in DFSORT
Application Programming Guide contains a sample ICETOOL job that shows how ICETOOL can be used to
perform complex tasks using multiple operations and data sets in a single step, along with a complete explanation
of the job.

Complete information on ICETOOL's messages and return codes can be found in DFSORT Messages, Codes and
Diagnosis Guide (SC26-7525).

Complete information on the newest features of ICETOOL can be found in the User Guide for DFSORT PTFs
UK90025 and UK90026 paper.

You can access all of these documents online by clicking the Publications link on the DFSORT home page at
URL:

http://www.ibm.com/storage/dfsort

Additional Sources for ICETOOL Examples
� The DFSORT home page on the World Wide Web at URL:

http://www.ibm.com/storage/dfsort

� DFSORT Application Programming Guide (SC26-7523)

� DFSORT: Getting Started (SC26-7527)

� z/OS DFSMSrmm Reporting (SC26-7406)

� z/OS DFSMShsm Data Recovery Scenarios (GC35-0419)

� z/OS SecureWay Security Server RACF Auditor's Guide (SA22-7684)

� z/OS SecureWay Security Server RACF Security Administrator's Guide (SA22-7683)

� RACFICE2 describes a technique for analyzing RACF data using ICETOOL. You can obtain RACFICE2 at:

www.ibm.com/systems/z/os/zos/features/racf/downloads/racfice.html

� The DFSORT product tape contains a set of illustrative examples of interest to Storage Administrators and
others who analyze data created by DFHSM, DFSMSrmm, DCOLLECT and SMF. The source for the fol-
lowing examples are available in sample job ICESTGEX:

– DCOLEX1 - DCOLLECT Example 1: VSAM report

– DCOLEX2 - DCOLLECT Example 2: Conversion reports

– DCOLEX3 - DCOLLECT Example 3: Capacity planning analysis and reports

– DFHSMEX1 - DFHSM Example 1: Deciphering Activity Logs

– DFHSMEX2 - DFHSM Example 2: Recover a DFHSM CDS with a broken index

– RMMEX1 - DFSMSrmm Example 1: SMF audit report

– RMMEX2 - DFSMSrmm Example 2: Create ADDVOLUME commands

ICESTGEX is also available via anonymous FTP from:

ftp.software.ibm.com/storage/dfsort/mvs/

What Can ICETOOL Do?

ICETOOL is a versatile DFSORT utility that allows you to perform multiple operations on one or more data sets in
a single job step.

ICETOOL uses the capabilities of DFSORT to perform the operations you request, calling DFSORT for each opera-
tion with the particular DFSORT control statements and options required.

The 17 ICETOOL operators, each of which can be used one or more times in a single run, allow you to perform a
variety of functions such as:

� Displaying statistical information for selected numeric fields, such as minimum, maximum, average, total, count
of values within a range and count of unique values. This makes it easy to extract frequently used analytical
data.

� Displaying list data sets showing character (up to 4000 bytes) and numeric fields in a variety of report formats,
allowing control of title, date, time, page numbers, carriage control characters, headings, lines per page, field
formats, and total, maximum, minimum, average and count values for the columns of numeric data. This
makes it easy to create simple, tailored and sectioned reports.

� Identifying and displaying invalid decimal values and their locations in a data set. This makes it easy to avoid
using invalid fields for other operations or steps.

� Creating multiple copies of sorted, merged, edited, or unedited data sets. This makes it easy to create several
identical data set copies.

� Sorting records between headers and trailers. This makes it easy to keep headers and trailers in place while
sorting the data records between them.

� Creating output data sets containing different subsets or field arrangements of input data sets. This makes it
easy to view data in many different ways.

� Creating a larger record from multiple shorter records, or multiple shorter records from a larger record. This
makes it easy to "resize" fixed length records.

� Displaying a list data set showing the DFSORT installation defaults selected at your site. This makes it easy to
determine the options selected and accepted for all eight installation environments.

� Creating output data sets and displaying list data sets for records with duplicate values, non-duplicate values, or
values that occur n times, less than n times, or more than n times. This makes it easy to view data according
to occurrences of values.

� Creating list data sets showing unique values for selected character and numeric fields and the number of times
each occurs, in a variety of report formats. This makes it easy to obtain reports based on occurrences of
values.

� Creating output data sets with information spliced together from two or more input records with duplicate
values. The information in the input records can originate from different data sets. This makes it easy to
perform various file "join" and "match" operations.

� Displaying the count of records in a data set, or creating an output data set containing an output record with
text and the count of records. This makes it easy to determine the number of records in a data set or in a
subset of a data set.

� Setting RC=12, RC=8, RC=4 or RC=0 based on the count of records in a data set. This makes it easy to avoid
using data sets that are empty, not empty or contain a certain number of records, for other operations or steps.

� Allowing operations to be performed or suppressed based on the success or failure of previous operations. This
makes it easy to group operations according to the action to be taken after an error.

You can use ICETOOL for SORT, COPY and MERGE applications.

ICETOOL can be called directly or from a program. Operators can be supplied in a data set. Alternatively,
operators can be supplied by a calling program parameter list, in which case ICETOOL returns information for each
operation in the parameter list. In either case, ICETOOL prints messages and gives a return code for each opera-
tion.

The 17 ICETOOL operators are: COPY, COUNT, DATASORT, DEFAULTS, DISPLAY, MERGE, MODE,
OCCUR, RANGE, RESIZE, SELECT, SORT, SPLICE, STATS, SUBSET, UNIQUE and VERIFY. By using com-
binations of these 17 operators, you can easily create applications that perform many complex tasks. By supplying
these operators in a parameter list, you can use the information returned by ICETOOL in your program.

General ICETOOL Job Format

Here's a representation of an ICETOOL job:

//EXAMP JOB ...
//TOOL EXEC PGM=ICETOOL
//TOOLMSG DD SYSOUT=A
//DFSMSG DD SYSOUT=A
//TOOLIN DD *
 <ICETOOL statements go here>
/*
<Additional JCL statements go here>

Required JCL Statements
� EXEC PGM=ICETOOL - to use ICETOOL directly.

� TOOLMSG - ICETOOL message data set. This data set has the same attributes as the DFSORT SYSOUT data
set.

� DFSMSG - DFSORT message data set. This data set has the same attributes as the DFSORT SYSOUT data
set.

� TOOLIN - ICETOOL statements. This data set has the same attributes as the DFSORT SYSIN data set.

� Additional JCL - as required for operators you specify.

ICETOOL Operator Statement Syntax

 Operator Statements
� Each ICETOOL operator statement describes a task you want ICETOOL to perform.

� Any number of operators can be specified and in any order.

� The general format for all ICETOOL statements is:

operator operand ... operand

Example:

COPY FROM(IN) TO(OUT1,OUT2)

� operator is one of the sixteen ICETOOL operator names.

� operand is keyword or keyword(parameter,...)

� One or more blanks can be used before the operator and between operands.

� Columns 1-72 are scanned; columns 73-80 are ignored.

� Continuation can be indicated by a dash (-) after the operator or any operand. Each operand must be com-
pletely specified on one line.

Example:

SORT FROM(IN1) -
 TO(OUT1,OUT2,OUT3) -
 USING(ABCD)

 Comment Statements
� Asterisk (*) in column 1 indicates a comment statement.

� Comment statements are printed with other ICETOOL statements, but otherwise ignored.

 Blank Statements
� Blank in columns 1-72 indicates a blank statement.

� Blank statements are ignored since ICETOOL prints blank lines where appropriate.

ICETOOL Return Codes

ICETOOL sets a return code for each operation it performs.

For the step, ICETOOL sets the return code to the highest operator return code.

The return codes are:

� 0 - Successful completion. No errors were detected.

� 4 - Successful completion. DFSORT detected one or more warning conditions, or RC4 was used for COUNT
and the criteria was met.

� 8 - Unsuccessful completion. RC8 was used for COUNT and the criteria was met.

� 12 - Unsuccessful completion. ICETOOL detected one or more errors, or RC12 was used (or defaulted) for
COUNT and the criteria was met.

� 16 - Unsuccessful completion. DFSORT detected one or more errors.

� 20 - Message data set error. The TOOLMSG DD statement was not present or the TOOLMSG data set was
not opened.

� 24 - Unsupported operating system. This operating system is not supported by this release of DFSORT.

ICETOOL Job with all Operators

Here's a sample ICETOOL job that briefly explains the function of each ICETOOL operator and shows a simple
example of its use. Later pages explain each operator in the job and its parameters (required and optional), and
show the TOOLMSG messages produced for each operator.

//EXAMP JOB ...
//* JCL REQUIRED FOR ALL ICETOOL JOBS
//TOOL EXEC PGM=ICETOOL
//TOOLMSG DD SYSOUT=A
//DFSMSG DD SYSOUT=A
//TOOLIN DD *
* ICETOOL Operators - brief explanations and simple examples

* COPY - copies a data set to one or more output data sets.
* Multiple output is handled using a single pass over
* the input.
* Example: copy the N data set to the DASD, PRINT and TAPE
* data sets.
COPY FROM(N) TO(DASD,PRINT,TAPE)

* COUNT - prints a message in TOOLMSG containing the count of
* records in a data set. Can also be used to create an output
* data set containing text and the count, or to set RC=12,
* RC=8, RC=4 or RC=0 based on the count of records in
* a data set.
* Example: print a count of the number of records in the
* IN2 data set.
 COUNT FROM(IN2)

* DATASORT - sorts data records between header and trailer records
* in a data set to an output data set.
* Example: sort the IN3 data set to the SRT1 data set keeping the
* header record and two trailer records in place. Uses the
* DFSORT SORT statement in the SRT1CNTL data set to sort the
* data records.
DATASORT FROM(IN3) TO(SRT1) HEADER TRAILER(2) USING(SRT1)

* DEFAULTS - prints the DFSORT installation defaults in a
* separate list data set.
* Example: print the JCL, INV, TSO, TSOINV, TD1, TD2,
* TD3 and TD4 installation defaults selected, in the
* INSTDEF data set
 DEFAULTS LIST(INSTDEF)

* DISPLAY - prints the values and characters of specific
* numeric and character fields in a separate list data set.
* Simple, tailored or sectioned reports can be produced.
* Maximums, minimums, totals, averages and counts can be
* produced.
* Example: print a tailored report showing values from the
* three IN2 data set ON fields, with title, heading lines,
* maximum, minimum and count, in the DOUT data set.
DISPLAY FROM(IN2) LIST(DOUT) BLANK -
TITLE('IN2 Data Set Report') PAGE DATE TIME -
HEADER('Store') ON(50,15,CH) -
HEADER('Profit/(Loss)') ON(26,8,ZD,E1) -
HEADER('Employees') ON(2,3,ZD) -
MAXIMUM('Largest') MINIMUM('Smallest') -
COUNT('Number of stores') EDCOUNT(U03)

* MERGE - merges one or more data sets to one or more
* output data sets.
* Example: merge the IN01, IN02 and IN03 data sets to
* the MRGOUT output data set using the DFSORT control
* statements in the MRG1CNTL data set.
MERGE FROM(IN01,IN02,IN03) TO(MRGOUT) USING(MRG1)

* MODE - specifies the error checking and actions after error
* detection to be performed for a group of operators.
* Example: continue processing operators whether or not an
* error is detected.
 MODE CONTINUE

* OCCUR - prints each unique value for specified numeric and
* character fields and how many times it occurs in a separate
* list data set. Simple or tailored reports can be produced.
* The values printed can be limited to those for which the
* value meets specified criteria (e.g. only duplicate values).
* Example: print a report showing values from the IN1 data set
* ON field, with the number of times each value occurs, in the
* LIST1 data set.
OCCUR FROM(IN1) LIST(LIST1) TITLE('Dep''t Counts') -

 ON(35,3,CH) ON(VALCNT)

* RANGE - prints a message in TOOLMSG containing the count
* of values in a specified range for a specified numeric
* field in a data set.
* Example: print a count of the values in the IN2 data set
* ON field that are higher than -50, but lower than +100.
RANGE FROM(IN2) ON(20,2,PD) HIGHER(-50) LOWER(100)

* RESIZE - creates larger records from smaller records, or
* smaller records from larger records.
* Example: create 40 byte output records from 10 byte input
* records.
RESIZE FROM(L10) TO(L40) TOLEN(40)

* SELECT - selects records from a data set for inclusion
* in an output data set based on meeting criteria for the
* number of times specified numeric or character field
* values occur (e.g. only duplicate values).
* Records that are not selected can be saved in a separate
* output data set.
* Example: selects records from the IN1 data set, for the
* SEL1 data set, whose ON field occurs only once (i.e, only
* records with no duplicate ON field values).
SELECT FROM(IN1) TO(SEL1) ON(30,2,PD) NODUPS

* SORT - sorts a data set to one or more output data sets.
* Multiple output is handled using a single pass over
* the input.
* Example: sort the IN1 data set to the OUT1 data set using
* the DFSORT control statements in the CTL1CNTL data set.
SORT FROM(IN1) TO(OUT1) USING(CTL1)

* SPLICE - splices together specified fields from records
* that have the same specified numeric or character field
* values (i.e., duplicate values), but different
* information. Specified fields from two or more records
* can be combined to create an output record. The fields
* to be spliced can originate from records in different
* data sets, so various "join" and "match" operations can
* be performed.
* Example: for ON fields that occur more than once (i.e.,
* duplicate ON field values) in the CONCT data set, splices
* the WITH field from the last duplicate record into the
* first duplicate record.
SPLICE FROM(CONCT) TO(COMBINE) ON(11,5,CH) WITH(41,20)

* STATS - prints messages in TOOLMSG containing the minimum,
* maximum, average, and total for specified numeric fields in
* a data set.
* Example: print the minimum, maximum, average and total
* values for the three VLRIN data set ON fields.
* For variable-length records, ON(VLEN) gives statistics
* about the length of the records.
STATS FROM(VLRIN) ON(VLEN) ON(12,2,ZD) ON(18,5,FS)

* SUBSET - selects records from a data set based on keeping or
* removing header records (the first n records), relative records,
* or trailer records (the last n records).
* Records that are not selected can be saved in a separate
* output data set.
* Example: Copies records 5, 21 and 31-33 from the
* IN1 data set to the SUB1 data set.
SUBSET FROM(IN1) TO(SUB1) INPUT KEEP -
RRN(5) RRN(21) RRN(31,33)

* UNIQUE - prints a message in TOOLMSG containing the count of
* unique values for a specified numeric or character field.
* Example: print the count of unique values in the
* OUT1 data set ON field.
UNIQUE FROM(OUT1) ON(30,2,PD)

* VERIFY - examines specified decimal fields in a data set and
* prints a message in TOOLMSG identifying each invalid value
* found for each field.
* Example: identify all values in the two IN2 data set
* decimal ON fields that have invalid digits (A-F)
* and/or invalid signs (0-9).
VERIFY FROM(IN2) ON(10,2,ZD) ON(41,6,PD)

/*
//* JCL REQUIRED FOR THE SPECIFIED OPERATORS
//N DD DSN=Y897797.NAMES,DISP=SHR
//DASD DD DSN=Y897797.OUT.COPY,DISP=OLD
//PRINT DD SYSOUT=A

//TAPE DD DSN=TAPE1,UNIT=3490,DISP=(,KEEP),VOL=SER=VOL001,
// LABEL=(,SL)
//INSTDEF DD SYSOUT=A
//IN2 DD DSN=Y897797.FIXED2.IN,DISP=SHR
//IN3 DD DSN=Y897797.FIXED3.IN,DISP=SHR
//L10 DD DSN=Y897797.FIXED.L10,DISP=SHR
//SRT1 DD DSN=&&SRT1,DISP=(,PASS),SPACE=(TRK,(5,5)),UNIT=SYSDA
//SRT1CNTL DD *
 SORT FIELDS=(11,10,CH,A)
/*
//DOUT DD SYSOUT=A
//IN1 DD DSN=Y897797.FIXED1.IN,DISP=SHR
//LIST1 DD SYSOUT=A
//SEL1 DD DSN=&&SEL1,DISP=(,PASS),SPACE=(TRK,(5,5)),UNIT=SYSDA
//L40 DD DSN=Y897797.FIXED.L40,DISP=OLD
//OUT1 DD DSN=&&TEMP,DISP=(,PASS),SPACE=(TRK,(5,5)),UNIT=SYSDA
//CTL1CNTL DD *
 SORT FIELDS=(35,3,CH,A)
 OMIT COND=(35,3,SS,EQ,C'J82,D54')
/*
//CONCT DD DSN=Y897797.BASE.IN,DISP=SHR
// DD DSN=Y897797.OVERLAY.IN,DISP=SHR
//COMBINE DD DSN=Y897797.COMBINE.OUT,DISP=(NEW,CATLG,DELETE),
// SPACE=(CYL,(5,5)),UNIT=SYSDA
//VLRIN DD DSN=Y897797.VARIABLE.IN,DISP=SHR
//SUB1 DD DSN=&&SUB1,DISP=(,PASS),SPACE=(TRK,(5,5)),UNIT=SYSDA
//IN01 DD DSN=Y897797.MERGEIN1,DISP=SHR
//IN02 DD DSN=Y897797.MERGEIN2,DISP=SHR
//IN03 DD DSN=Y897797.MERGEIN3,DISP=SHR
//MRGOUT DD SYSOUT=A
//MRG1CNTL DD *
 OPTION EQUALS
 MERGE FIELDS=(21,4,CH,A)
/*

COPY Operator Details

 Syntax
COPY FROM(indd) TO(outdd,...) USING(xxxx) VSAMTYPE(x)
 JKFROM

 LOCALE(name) SERIAL
 LOCALE(CURRENT)
 LOCALE(NONE)

 Function

Copies the indd data set to the outdd data sets (up to 10). The DFSORT control statements in xxxxCNTL are used
if USING(xxxx) is specified. DFSORT control statements and options can be used to copy a subset of the input
records (INCLUDE or OMIT statement; SKIPREC and STOPAFT options; OUTFIL INCLUDE, OMIT, SAVE,
STARTREC, ENDREC, ACCEPT, SAMPLE, SPLIT, SPLITBY and SPLIT1R operands), reformat records for
output (INREC, OUTREC and OUTFIL), and so on. See DFSORT Application Programming Guide for complete
details of DFSORT control statements and options.

 Examples
COPY FROM(IN1) TO(NEW,BACKUP) USING(CTL1)

COPY FROM(VSAMIN) TO(VSAMOUT) VSAMTYPE(V)

COPY FROM(MASTER) USING(OUTF)

COPY JKFROM TO(JOUT) USING(CTL2)

CTL1CNTL might contain the following:

//CTL1CNTL DD *
 OMIT COND=(28,5,PD,NE,NUM)
/*

Note the use of DFSORT's non-numeric test feature to OMIT records with invalid PD values. Complete details on
this feature of INCLUDE/OMIT can be found in DFSORT Application Programming Guide.

OUTFCNTL might contain the following:

//OUTFCNTL DD *
 OUTFIL FNAMES=FMT1,BUILD=(C'SUB1',32,15,2Z,1,3,PD)
 OUTFIL FNAMES=FMT2,BUILD=(C'SUB2',16,2,BI,M1,C'**',32,15)
/*

Note that DFSORT's OUTFIL BUILD feature is used here to create two output data sets with different field
arrangements using a single pass over the input data set. Complete details on this feature of OUTFIL as well as its
many other features can be found in DFSORT Application Programming Guide.

CTL2CNTL might contain the following:

//CTL2CNTL DD *
 JOINKEYS F1=IN1,FIELDS=(5,12,A)
 JOINKEYS F2=IN2,FIELDS=(11,12,A)
 REFORMAT FIELDS=(F1:4,40,F2:15,20)
/*

Note the use of JKFROM to indicate a JOINKEYS application for the COPY operator (JOINKEYS and
REFORMAT in CTL2CNTL). Complete details on JOINKEYS applications can be found in DFSORT Application
Programming Guide.

 Required Operands
� FROM or JKFROM

– FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

– JKFROM - used for a JOINKEYS application. You must provide a USING(xxxx) operand. In
xxxxCNTL, you must provide a JOINKEYS statement with F1=ddname1 for the F1 file and a JOINKEYS
statement with F2=ddname2 for the F2 file, as well as JOIN and REFORMAT statements as needed.

� TO - the ddnames of 1 to 10 output data sets. You must supply DD statements for the ddnames you specify.
TO, USING, or TO and USING must be specified.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL if you specify USING(xxxx). TO, USING, or TO and USING
must be specified.

 Optional Operands
� VSAMTYPE - the record format for a VSAM input data set (F or V).

� LOCALE - overrides the installation default for locale processing.

� SERIAL - causes OUTFIL processing not to be used for multiple outdd data sets. SERIAL is not recom-
mended because it imposes data set restrictions and can degrade performance.

 Note
� If you use OUTFIL statements in xxxxCNTL to specify your output data sets, you do not need to specify TO.

Example of TOOLMSG Output for COPY
* COPY - copies a data set to one or more output data sets.
* Multiple output is handled using a single pass over
* the input.
* Example: copy the N data set to the DASD, PRINT and TAPE
* data sets.
COPY FROM(N) TO(DASD,PRINT,TAPE)

ICE627I 0 DFSORT CALL 0001 FOR COPY FROM N TO OUTFIL COMPLETED
ICE602I 0 OPERATION RETURN CODE: 00

Note that DFSORT's OUTFIL multiple output feature is used automatically to create the three output data sets with
a single pass over the input data set.

COUNT Operator Details

 Syntax
COUNT FROM(indd) USING(xxxx) VSAMTYPE(x) LOCALE(name)
 LOCALE(CURRENT)
 LOCALE(NONE)

EMPTY RC4 SUB(q) WRITE(countdd) TEXT('string')
 NOTEMPTY RC8 ADD(r)
 HIGHER(x) RC12
 LOWER(y)
 EQUAL(v)
 NOTEQUAL(w)

 DIGITS(d) WIDTH(n)
 EDCOUNT(formatting)

 Function

Prints a message in TOOLMSG containing the count of records in the indd data set. The DFSORT control state-
ments in xxxxCNTL are used if USING(xxxx) is specified. The INCLUDE or OMIT statement can be used to
count a subset of the input records.

COUNT can also be used to:

� subtract a value from the count

� add a value to the count

� create an output data set containing text and the count, or just the count

� set RC=12, RC=8 or RC=4 if the record count meets specified criteria (that is, empty, not empty, n records, not
n records, more than n records or less than n records), or RC=0 if the record count does not meet specified
criteria.

 Examples
COUNT FROM(IN1)

COUNT FROM(INPUT5) EMPTY

COUNT FROM(MASTER) HIGHER(50000) RC4

COUNT FROM(UPDATE2) NOTEMPTY SUB(2)

COUNT FROM(IN1) USING(PDAT) WRITE(CT) TEXT('Count is') -
 DIGITS(6)

PDATCNTL might contain the following:

//PDATCNTL DD *
 INCLUDE COND=(15,7,CH,EQ,DATE3-1)
/*

Note the use of DFSORT's past date feature to INCLUDE records with yesterday's date in the form yyyyddd.
Complete details on this feature of INCLUDE/OMIT can be found in DFSORT Application Programming Guide.

 Required Operand
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

 Optional Operands
� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You

must supply a DD statement for xxxxCNTL if you specify USING(xxxx).

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� LOCALE - overrides the installation default for locale processing.

� EMPTY, NOTEMPTY, HIGHER, LOWER, EQUAL, NOTEQUAL - defines the criteria against which the
record count is to be matched. If the criteria is met, ICETOOL sets RC=4 if RC4 is specified, RC=8 if RC8 is
specified, or RC=12 if RC12 is specified or by default. If the criteria is not met, ICETOOL sets RC=0 for this
COUNT operator. x, y, v, and w must be specified as n or +n where n can be 0 to 562949953421310.

� RC4 - sets RC=4 if the criteria is met (overriding the default of RC=12).

� RC8 - sets RC=8 if the criteria is met (overriding the default of RC=12).

� RC12 - sets RC=12 if the criteria is met (same as the default of RC=12).

� SUB - subtracts a value from the record count. If SUB reduces the record count below 0, the record count is
set to 0. q must be specified as n or +n where n can be 1 to 999.

� ADD - adds a value to the record count. r must be specified as n or +n where n can be 1 to 999.

� WRITE - the ddname of the count data set. You must supply a DD statement for the ddname you specify.

� TEXT - a string to be printed before the count in the count data set (overriding the default of just printing the
count). The string can be 1 to 50 characters.

� DIGITS - the number of digits for the count in the count data set (overriding the default of 15). d can be 1 to
15.

� EDCOUNT - specifies how the count is to be formatted for printing. The following formatting items can be
used (see DFSORT Application Programming Guide for complete details): mask, E'pattern', L'string', F'string',
T'string', LZ and Udd.

� WIDTH - the record length and LRECL for the count data set. n can be 1 to 32760. If the WIDTH specified
is greater than the calculated record length, ICETOOL pads the count record on the right with blanks to the
specified record length. If the WIDTH specified is less than the calculated record length, ICETOOL terminates
the operation. If WIDTH is not specified, the calculated record length is used as the record length and LRECL
for the count data set.

 Symbols
� A symbol for a character constant can be used instead of 'string' in the TEXT operand.

� Symbols for decimal constants can be used instead of n and +n in the HIGHER, LOWER, EQUAL,
NOTEQUAL, SUB and ADD operands.

 Notes
� The record count is also printed for the DISPLAY, OCCUR, RANGE, SELECT, STATS, UNIQUE, and

VERIFY operators.

� You can use COUNT to set RC=12 or RC=8 based on the record count to stop other operations or steps from
executing. When you specify one of the criteria operands, ICETOOL does not print the record count, and it
uses DFSORT's STOPAFT option to process the minimum number of records required to determine whether or
not the criteria is met.

� RECFM=FB is used for the count data set.

� SUB(q) and ADD(r) are useful for input data sets with header or trailer records.

Example of TOOLMSG Output for COUNT
* COUNT - prints a message in TOOLMSG containing the count of
* records in a data set. Can also be used to create an output
* data set containing text and the count, or to set RC=12,
* RC=8, RC=4 or RC=0 based on the count of records in
* a data set.
* Example: print a count of the number of records in the
* IN2 data set.

 COUNT FROM(IN2)
ICE627I 0 DFSORT CALL 0002 FOR COPY FROM IN2 TO E35 EXIT COMPLETED
ICE628I 0 RECORD COUNT: 000000000000008
ICE602I 0 OPERATION RETURN CODE: 00

Example of Using COUNT to Set RC=12 or RC=0
COUNT FROM(IN5) EMPTY

TOOLMSG Output if IN5 is empty

ICE646A 0 RECORD COUNT MEETS CRITERIA - RC=12 SET
ICE602I 0 OPERATION RETURN CODE: 12

TOOLMSG Output if IN5 is not empty

ICE647I 0 RECORD COUNT DOES NOT MEET CRITERIA - RC=0 SET
ICE602I 0 OPERATION RETURN CODE: 00

Example of Using COUNT to Set RC=4 or RC=0
COUNT FROM(IN5) HIGHER(5) RC4

TOOLMSG Output if IN5 has 6 or more records

ICE651A 0 RECORD COUNT MEETS CRITERIA - RC=4 SET
ICE602I 0 OPERATION RETURN CODE: 04

TOOLMSG Output if IN5 has 5 or less records

ICE647I 0 RECORD COUNT DOES NOT MEET CRITERIA - RC=0 SET
ICE602I 0 OPERATION RETURN CODE: 00

DATASORT Operator Details

 Syntax
DATASORT FROM(indd) TO(outdd) USING(xxxx)

HEADER TRAILER VSAMTYPE(x)
 FIRST LAST
 HEADER(u) TRAILER(v)
 FIRST(u) LAST(v)

 Function

Copies one or more header records and/or one or more trailer records to the outdd data set in their original indd
record order, while sorting the indd data records between the header and trailer records to the outdd data set. Thus,
the first n records (header records) and/or last n records (trailer records) are kept in place and the data records
between them are sorted.

You must supply a DFSORT SORT statement in xxxxCNTL to indicate the control fields for sorting the data
records. Additional DFSORT control statements in xxxxCNTL are used if specified.

 Examples
DATASORT FROM(INPUT) TO(OUTPUT) HEADER TRAILER USING(CTL1)

DATASORT FROM(FILE1) TO(FILE2) LAST(5) USING(CTL1)

DATASORT FROM(IN1) TO(OUT1) HEADER(3) TRAILER(2) USING(CTL1)

CTL1CNTL might contain the following:

 SORT FIELDS=(11,5,CH,A,21,6,SFF,D)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� TO - the ddname of the output data set for the sorted records. You must supply a DD statement for the
ddname you specify.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL. You must supply a DFSORT SORT statement in xxxxCNTL.
See z/OS DFSORT Application Programming Guide for details of the other DFSORT control statements you
can use with SUBSET.

� HEADER, FIRST, TRAILER, LAST - defines the number of header records (first u records) and/or trailer
records (last v records) to be kept in place. u and v must be specified as n or +n where n can be 1 to 1000000.

 Optional Operands
� VSAMTYPE - the record format for a VSAM input data set (F or V).

 Symbols
� Symbols for decimal constants can be used instead of n and +n in the HEADER, FIRST, TRAILER and LAST

operands.

 Notes
� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort performed for the

DATASORT operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for DATASORT
* DATASORT - sorts data records between header and trailer records
* in a data set to an output data set.
* Example: sort the IN3 data set to the SRT1 data set keeping the
* header record and two trailer records in place. Uses the
* DFSORT SORT statement in the SRT1CNTL data set to sort the
* data records.
DATASORT FROM(IN3) TO(SRT1) HEADER TRAILER(2) USING(SRT1)

ICE606I 0 DFSORT CALL 0003 FOR SORT FROM IN3 TO SRT1 USING SRT1CNTL COMPLETED
ICE602I 0 OPERATION RETURN CODE: 00

DEFAULTS Operator Details

 Syntax
DEFAULTS LIST(listdd) LISTSDB
 LISTNOSDB

 Function

Prints, in the listdd data set, the DFSORT installation defaults. DFSORT lets you maintain eight separate sets of
installation defaults for eight different installation environments as follows:

� JCL (ICEAM1) - batch JCL directly invoked installation environment

� INV (ICEAM2) - batch program invoked installation environment

� TSO (ICEAM3) - TSO directly invoked installation environment

� TSOINV (ICEAM4) - TSO program invoked installation environment

� TD1 (ICETD1) - first time-of-day installation environment

� TD2 (ICETD2) - second time-of-day installation environment

� TD3 (ICETD3) - third time-of-day installation environment

� TD4 (ICETD4) - fourth time-of-day installation environment

Each installation default has two or more possible values; DFSORT is shipped with a set of IBM-supplied defaults
that can be modified using ICEPRMxx members of PARMLIB or the ICEMAC macro. The DEFAULTS operator
provides an easy way to determine the installation default values to be used at run-time.

DEFAULTS produces a three-part report showing:

1. The merged PARMLIB/ICEMAC installation default values for ICEAM1-4 and ICETD1-4 that will be used at
run-time.

2. The specified PARMLIB ICEPRMxx member option values for ICEAM1-4 and ICETD1-4 (for reference).

3. The ICEMAC installation default values for ICEAM1-4 and ICETD1-4 (for reference).

See z/OS DFSORT Installation and Customization for a complete discussion of ICEPRMxx members in PARMLIB,
the ICEMAC macro, the eight installation environments, the installation default values and the DEFAULTS report.

 Examples
DEFAULTS LIST(SHOWDEF)

DEFAULTS LIST(DFLTS) LISTSDB

 Required Operand
� LIST - the ddname of the list data set. You must supply a DD statement for the ddname you specify.

 Optional Operands
� LISTSDB - uses SDB for the list data set (overrides installation value SDBMSG=NO).

� LISTNOSDB - does not use SDB for the list data set (overrides installation value SDBMSG=YES).

 Notes
� The control character occupies the first byte of each record in the list data set.

� RECFM=FBA and LRECL=121 are used for the list data set.

Example of TOOLMSG Output for DEFAULTS
* DEFAULTS - prints the DFSORT installation defaults in a
* separate list data set.
* Example: print the JCL, INV, TSO, TSOINV, TD1, TD2,
* TD3 and TD4 installation defaults selected, in the
* INSTDEF data set

 DEFAULTS LIST(INSTDEF)
ICE603I 0 INFORMATION PRINTED IN INSTDEF DATA SET
ICE602I 0 OPERATION RETURN CODE: 00

Example of List Output for DEFAULTS

Here's an example of the output that would appear in INSTDEF:

Z/OS DFSORT V1R12 MERGED PARMLIB/ICEMAC DEFAULTS - 1 -

* IBM-SUPPLIED DEFAULT (ONLY SHOWN IF DIFFERENT FROM THE
 SPECIFIED DEFAULT)

ITEM JCL (ICEAM1) INV (ICEAM2) ...
---------- -------------------- -------------------- ...
ENABLE NONE TD1 ...

ABCODE MSG 99 ...
 * MSG ...
ALTSEQ SEE BELOW SEE BELOW ...
ARESALL 0 0 ...
ARESINV NOT APPLICABLE 0 ...
CFW YES YES ...
CHALT YES YES ...
 * NO * NO ...
. . . ...
. . . ...
. . . ...

  ICETOOL Mini-User Guide 19



  
 

DISPLAY Operator Details

 Syntax
DISPLAY FROM(indd) LIST(listdd) ON(p,m,f) ...
 ON(p,m,f,formatting)
 ON(p,m,HEX)
 ON(VLEN)
 ON(VLEN,formatting)
 ON(NUM)
 ON(NUM,formatting)

TITLE('string1') ... TLEFT TFIRST
 TITLE('string1','string2')
 TITLE('string1','string2','string3')

 PAGE DATE TIME NOCC
 DATE(abcd) TIME(abc)
 DATENS(abc) TIMENS(ab)
 YDDD(abc)
 YDDDNS(ab)

HEADER('string1') ... LINES(n) BLANK
 HEADER('string1','string2') PLUS
 HEADER('string1','string2','string3')
 HEADER(NONE)
 NOHEADER

TOTAL('string') MAXIMUM('string') MINIMUM('string')

AVERAGE('string') COUNT('string') EDCOUNT(formatting)

LIMIT(n) VSAMTYPE(x) WIDTH(n)

 BREAK(p,m,f) BTITLE('string') BTOTAL('string')
 BREAK(p,m,f,formatting)

BMAXIMUM('string') BMINIMUM('string') BAVERAGE('string')

 BCOUNT('string') EDBCOUNT(formatting)

INDENT(n) BETWEEN(n) TBETWEEN(n) STATLEFT UZERO LISTSDB

 Function

Prints, in the listdd data set, the values or characters of specified numeric and/or character fields. The fields are
printed in columns in the same order in which they are specified. From 1 to 50 ON fields can be specified as long
as the resulting list data set line length does not exceed the limit specified by the WIDTH operand or 8192 bytes.

Simple, tailored or sectioned reports can be produced:

� Up to three title lines can appear at the top of each page, or only at the top of the first page. The first title line
can consist of a title string, the page number, the date and/or the time, in any order specified. The second title
line consists of a specified title string. The third title line consists of a specified title string. The title strings
can be centered or left-justified with respect to each other.

� Carriage control characters can be printed or suppressed.

� One, two or three line headings can be printed to identify each specified field.

� Overall statistics (total, maximum, minimum, average, and/or count) can be produced.

� Column widths are adjusted automatically according to the operands specified and the length of the fields.

� Alternate formats for date and time can be specified.

� Numeric fields, consisting of a sign and up to 31 digits in various formats, and counts can be edited with
separators, decimal points, patterns, division, and leading, trailing and floating signs.

� A break field can be used to produce sections with their own break title and break statistics (total, maximum,
minimum, average and/or count).

 Examples
DISPLAY FROM(DATA) LIST(SHOWIT) ON(10,44,CH) ON(5,4,FS)

DISPLAY FROM(VLR) LIST(RDWLIST) -
INDENT(10) BETWEEN(5) TBETWEEN(4) -
TITLE('Record Length Report') DATE -
TITLE('for VB data set') -
HEADER('Record','number') ON(NUM,U04) -
HEADER('Record','length') ON(VLEN,U05) -
HEADER(,'RDW in Hex') ON(1,4,HEX) -
TOTAL('Total') AVERAGE('Average') -
MINIMUM('Minimum') MAXIMUM('Maximum') -
COUNT('Number of records ') EDCOUNT(U06)

DISPLAY FROM(MASTER) LIST(DENMARK) LINES(50) -
PAGE TITLE('Report for Denmark') DATE(DM4-) -
BTITLE('Division:') BREAK(23,10,CH) -
HEADER('Part') ON(15,6,CH) -
HEADER('Completed') ON(3,4,ZD,A2) -
HEADER('Value (kr)') ON(38,12,SFF,C2) -
BMINIMUM('Lowest in this Division:') -
BMAXIMUM('Highest in this Division:') -
BAVERAGE('Average in this Division:') -
BTOTAL('Total for this Division:') -
AVERAGE('Average for all Divisions:')

DISPLAY FROM(DATA1) LIST(KBDATA) WIDTH(150) -
NOHEADER NOCC ON(15,44,CH) ON(8,4,PD,/KB,T' KB')

DISPLAY FROM(SMF14) LIST(SMF14RPT) -
TITLE('SMF Type-14 Records') DATENS(4MD) -

 LISTSDB -
HEADER('Date') ON(11,4,DT1,E'9999/99/99') -
HEADER('Time') ON(7,4,TM1,E'99:99:99') -
HEADER('Sys') ON(15,4,CH) -
HEADER('Jobname') ON(19,8,CH) -

 HEADER('Datasetname') ON(69,44,CH)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� LIST - the ddname of the list data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation. From 1 to 50 ON fields can be specified.

– (p,m,f) gives the position, length and format of a numeric or character field. A field must not extend
beyond position 32752 or the end of the record.

– (p,m,f,formatting) gives the position, length and format of a numeric or character field and specifies how
the data for this field is to be formatted for printing. A field must not extend beyond position 32752 or the
end of the record.

The following formatting items can be used (see DFSORT Application Programming Guide for complete
details):

— Mask - an edit mask to be applied to the numeric data. Thirty-nine pre-defined edit masks are avail-
able, encompassing many of the numeric notations throughout the world with respect to separators,
decimal point, decimal places, signs and so on. See "Appendix A" for complete descriptions and
examples of all thirty-nine masks. The attributes of each group of masks is shown below.

— E'pattern' - an edit pattern (1 to 44 characters) to be applied to the numeric data. Each 9 in the pattern
(up to 31) is replaced by a corresponding digit (0-9) from the numeric value. Other characters in the
pattern appear as specified. E'pattern' can be used for formatting unsigned numeric data such as tele-
phone numbers, dates, and so on.

— L'string' - a leading string to appear at the beginning of the character or numeric data column. Each
string can be 1 to 10 characters.

— F'string' - a floating string to appear to the left of numeric data. Each string can be 1 to 10 characters.

— T'string' - a trailing string to appear at the end of the character or numeric data column. Each string
can be 1 to 10 characters.

— LZ - used with an edit mask to print leading zeros for the numeric field (overriding the default of
suppressing leading zeros).

— NOST - used to suppress specified statistics for the numeric field (overriding the default of printing
specified statistics).

— Ndd - use dd digits for the numeric field where dd is greater than the calculated number of digits. dd
must be a two-digit number from 01 to 31.

 
Masks

 
Separators

Decimal
Places

Positive
Sign

Negative
Sign

A0 No 0 blank -

A1-A5 Yes 0 blank -

B1-B6 Yes 1 blank -

C1-C6 Yes 2 blank -

D1-D6 Yes 3 blank -

E1-E4 Yes 0 blank ( )

F1-F5 Yes 2 blank ( )

G1-G6 Yes 4 blank -

— Udd - use dd digits for the numeric field where dd is greater than or less than the calculated number
of digits. dd must be a two-digit number from 01 to 31.

— /x - used to divide numeric data before formatting. /x indicates a division factor as follows:

 � /D: 10

 � /C: 100

 � /K: 1000

 � /KB: 1024

 � /DK: 10*1000

 � /CK: 100*1000

 � /M: 1000*1000

 � /MB: 1024*1024

 � /G: 1000*1000*1000

 � /GB: 1024*1024*1024

– (p,m,HEX) gives the position and length of a character field to be printed in hexadecimal format. A field
must not extend beyond position 32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record. The following formatting items can be used with VLEN: mask, E'pattern', L'string', F'string',
T'string', LZ, NOST, Ndd, Udd and /x.

– NUM specifies that the record number is to be printed starting at 1 and incrementing by 1 for each record.
The following formatting items can be used with NUM: mask, E'pattern', L'string', F'string', T'string', LZ,
Ndd and Udd.

– A description of each type of field is given below.

Format Code Length Description

BI 1-8 bytes Unsigned binary

FI 1-8 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CH 1-4000 bytes Character

FL 4 or 8 bytes Signed hexadecimal floating-point as integer

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

DT1 4 bytes SMF date as Z'yyyymmdd'

DT2 4 bytes SMF date as Z'yyyymm'

DT3 4 bytes SMF date as Z'yyyyddd'

DC1 8 bytes TOD date as Z'yyyymmdd'

DC2 8 bytes TOD date as Z'yyyymm'

DC3 8 bytes TOD date as Z'yyyyddd'

 Optional Operands
� TITLE, PAGE, DATE, DATENS, YDDD, YDDDNS, TIME and TIMENS - the elements to appear in the title

lines at the top of each page. Only specified elements appear and in the order given. Each title string can
consist of up to three individual strings. The page, date and time appear in the first title line with the first title
string. The second title string appears in the second title line. The third title string appears in the third title
line. A string can be 1 to 50 characters; the maximum length for each title string can be 50 characters.
Alternate formats for date and time can be specified.

� TLEFT - title strings are left-justified with respect to each other (overriding the default of centering title strings
with respect to each other).

� TFIRST - the title lines are only printed on the first page (overriding the default of printing the title lines on
every page).

� NOCC - suppresses carriage control characters (overriding the default of printing carriage control characters).

� HEADER - a one, two or three line heading to be printed for the corresponding ON field (overriding the
"standard" one line heading of "(p,m,f)"). A heading string can be 1 to 50 characters. NONE can be used to
suppress the heading for the corresponding ON field.

� NOHEADER - suppresses the heading line.

� LINES - the number of lines per page (overriding the default of 58). n can be 10 to 999.

� BLANK and PLUS - causes the column widths to be dynamically adjusted as needed (overriding the standard
fixed column widths) and suppresses leading zeros for numeric fields. BLANK causes a blank, rather than a +
to be used for the positive sign.

Format Code Length Description

DE1 8 bytes ETOD date as Z'yyyymmdd'

DE2 8 bytes ETOD date as Z'yyyymm'

DE3 8 bytes ETOD date as Z'yyyyddd'

TM1 4 bytes SMF time as Z'hhmmss'

TM2 4 bytes SMF time as Z'hhmm'

TM3 4 bytes SMF time as Z'hh'

TM4 4 bytes SMF time as Z'hhmmssxx'

TC1 8 bytes TOD time as Z'hhmmss'

TC2 8 bytes TOD time as Z'hhmm'

TC3 8 bytes TOD time as Z'hh'

TC4 8 bytes TOD time as Z'hhmmssxx'

TE1 8 bytes ETOD time as Z'hhmmss'

TE2 8 bytes ETOD time as Z'hhmm'

TE3 8 bytes ETOD time as Z'hh'

TE4 8 bytes ETOD time as Z'hhmmssxx'

HEX 1-2000 bytes Character printed as hexadecimal

VLEN n/a Record length for VLR (1,2,BI)

NUM n/a Relative record number

� TOTAL, MAXIMUM, MINIMUM, AVERAGE and COUNT - the overall statistics to appear after the columns
of data for the report. Only the specified statistics appear and in the order given. Each string can be 1 to 50
characters.

� EDCOUNT - specifies how the overall count is to be formatted for printing. The following formatting items
can be used (see DFSORT Application Programming Guide for complete details): mask, E'pattern', L'string',
F'string', T'string', LZ and Udd.

� LIMIT - a limit for the number of invalid decimal values (overriding the default of 200). If n invalid decimal
values are for an explanation of invalid decimal values.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� WIDTH - the line length and LRECL for the list data set. n can be 121 to 8192 (or 121 to 8191 if NOCC is
specified). If the WIDTH specified is less than the calculated line length, ICETOOL terminates the operation.
If WIDTH is not specified, the calculated line length (subject to a minimum of 121, or 120 if NOCC is speci-
fied), is used as the line length and LRECL for the list data set.

� LONGLINE - if WIDTH(n) is not specified, allows the calculated line length to be up to 8192 bytes (over-
riding the default of 2048 bytes).

� BREAK - the break field to be used to divide the report into sections. Each set of sequential input records
with the same break field value is treated as a section in the report. Each section starts on a new page with its
own section title and section statistics. The following formatting items can be used with BREAK: mask,
E'pattern', L'string', F'string', T'string', LZ and Udd.

� BTITLE - a string to appear in the section title. The break field and string appear in the section title in the
order given. The string can be 1 to 50 characters.

� BTOTAL, BMAXIMUM, BMINIMUM, BAVERAGE and BCOUNT - the section statistics to appear after the
columns of data for each section. Only the specified statistics appear for each section and in the order given.
Each string can be 1 to 50 characters.

� EDBCOUNT - specifies how the section count is to be formatted for printing. The following formatting items
can be used (see DFSORT Application Programming Guide for complete details): mask, E'pattern', L'string',
F'string', T'string', LZ and Udd.

� INDENT - the number of blanks used to indent the report (overriding the default of 0). n can be 0 to 50.

� BETWEEN - the number of blanks used between the columns of data (overriding the default of 3). n can be 0
to 50.

� TBETWEEN - the number of blanks used between title elements (overriding the default of 8). n can be 0 to
50.

� STATLEFT - places the statistics strings to the left of the first column of data (overriding the default of
placing the strings in the first column).

� UZERO - causes -0 to be treated as unsigned, that is, as +0.

� LISTSDB - uses SDB for the list data set (overrides installation value SDBMSG=NO).

� LISTNOSDB - does not use SDB for the list data set (overrides installation value SDBMSG=YES).

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON and BREAK operands.

� Symbols for character constants can be used instead of 'string' in the TITLE, HEADER, TOTAL, MAXIMUM,
MINIMUM, AVERAGE, COUNT, BTITLE, BTOTAL, BMAXIMUM, BMINIMUM, BAVERAGE and
BCOUNT operands.

 Notes
� The control character occupies the first byte of each record in the list data set unless NOCC is specified.

� By default, the first column starts in the second byte of each record in the list data set (or in the first byte if
NOCC is specified). INDENT(n) or STATLEFT can be used to override this default.

� By default, three blanks appear between columns in the list data set. BETWEEN(n) can be used to override
this default.

� By default, eight blanks appear between title elements in the list data set. TBETWEEN(n) can be used to
override this default.

� RECFM=FBA is used for the list data set if NOCC is not specified. RECFM=FB is used for the list data set if
NOCC is specified.

� Specifying LONGLINE can enable you to include more fields in your report, up to a maximum of 50.

Example of TOOLMSG Output for DISPLAY
* DISPLAY - prints the values and characters of specific
* numeric and character fields in a separate list data set.
* Simple, tailored or sectioned reports can be produced.
* Maximums, minimums, totals, averages and counts can be

 * produced.
* Example: print a tailored report showing values from the
* three IN2 data set ON fields, with title, heading lines,
* maximum, minimum and count, in the DOUT data set.
DISPLAY FROM(IN2) LIST(DOUT) BLANK -
TITLE('IN2 Data Set Report') PAGE DATE TIME -
HEADER('Store') ON(50,15,CH) -
HEADER('Profit/(Loss)') ON(26,8,ZD,E1) -
HEADER('Employees') ON(2,3,ZD) -
MAXIMUM('Largest') MINIMUM('Smallest') -
COUNT('Number of stores') EDCOUNT(U03)

ICE627I 0 DFSORT CALL 0004 FOR COPY FROM IN2 TO E35 EXIT COMPLETED
ICE603I 0 INFORMATION PRINTED IN DOUT DATA SET
ICE628I 0 RECORD COUNT: 000000000000008
ICE602I 0 OPERATION RETURN CODE: 00

Example of List Output for DISPLAY

Here's an example of the output that would appear in DOUT:

IN2 Data Set Report - 1 - 02/28/05 13:15:19

Store Profit/(Loss) Employees
--------------- ------------- ---------
San Jose 72,345,678 123
Morgan Hill 10,273 71
Palo Alto (52,766,111) 101
Sunnyvale 92,378,566 166
San Francisco 12 27
San Diego (64,832,715) 102
Los Altos 43,343,732 123
Gilroy 89,348,399 152

Largest 92,378,566 166

Smallest (64,832,715) 27

Number of stores 8

Example using OUTFIL Lookup and Change with DISPLAY
COPY FROM(IN) USING(OUTF)
DISPLAY FROM(TEMP) LIST(EMPCT) BLANK -

TITLE('Employees by Function') -
 YDDD(4D-) -

HEADER(,'Function') HEADER('Number of','Employees') -
 ON(1,25,CH) ON(30,4,ZD)

OUTFCNTL contains:

 OUTFIL FNAMES=TEMP,
 BUILD=(1:9,2,CHANGE=(25,
 C'MN',C'Manufacturing',

C'RD',C'Research and Development',
 C'FN',C'Finance',
 C'MR',C'Marketing',
 C'IS',C'Information Systems'),
 30:4,4)

Example of Lookup and Change with DISPLAY Output

Below is an example of the output that would appear in EMPCT.

 Employees by Function 2005-059

 Number of
 Function Employees
 ------------------------- ---------
 Manufacturing 486
 Marketing 21
 Research and Development 55
 Information Systems 123
 Finance 33

Note that the 2-character division codes at position 9 have been replaced by meaningful phrases using DFSORT's
OUTFIL lookup and change feature. Complete details on this feature of OUTFIL as well as its many other features
can be found in DFSORT Application Programming Guide.

Example of Plain and Fancy Reports
* Produce a plain report
 DISPLAY FROM(ACCTS) LIST(PLAIN) -

TITLE('Accounts Report for First Quarter') -
DATE(MD4/) BLANK -
HEADER('Amount') ON(12,6,ZD) -
HEADER('Id') ON(NUM) -
HEADER('Acct#') ON(31,3,PD) -
HEADER('Date') ON(1,4,ZD) -
TOTAL('Total for Q1') -
AVERAGE('Average for Q1')

* Produce a fancy report by using INDENT,
* BETWEEN, STATLEFT and formatting items
* to improve its appearance
 DISPLAY FROM(ACCTS) LIST(FANCY) TBETWEEN(16) -

TITLE('Accounts Report for First Quarter') -
DATE(MD4/) BLANK -
HEADER('Amount') ON(12,6,ZD,C1,N08) -
HEADER('Id') ON(NUM,N02) -
HEADER('Acct#') ON(31,3,PD,NOST,LZ) -
HEADER('Date') ON(1,4,ZD,E'99/99',NOST) -
INDENT(2) BETWEEN(5) -

 STATLEFT -
TOTAL('Total for Q1') -
AVERAGE('Average for Q1')

Example of Plain Report Output
Accounts Report for First Quarter 04/17/2006

 Amount Id Acct# Date
---------------- --------------- ---------------- ----------------
 93271 1 15932 106

137622 2 187 128
 83147 3 15932 212
 183261 4 2158 217

76389 5 187 305
 920013 6 15932 319

Total for Q1
 1493703 50328 1287

Average for Q1
 248950 8388 214

Example of Fancy Report Output
Accounts Report for First Quarter 04/17/2006

Amount Id Acct# Date
----------- --- ------ -----

 932.71 1 15932 01/06
 1,376.22 2 00187 01/28
 831.47 3 15932 02/12
 1,832.61 4 02158 02/17
 763.89 5 00187 03/05
 9,200.13 6 15932 03/19

Total for Q1 14,937.03

Average for Q1 2,489.50

MERGE Operator Details

 Syntax
MERGE FROM(indd,...) ... USING(xxxx) TO(outdd,...) VSAMTYPE(x)

 LOCALE(name) SERIAL
 LOCALE(CURRENT)
 LOCALE(NONE)

 Function

Merges the indd data sets (up to 50 in up to 10 FROM operands) to the outdd data sets (up to 10) using the
DFSORT control statements in xxxxCNTL. You must supply a DFSORT MERGE statement in xxxxCNTL to
indicate the control fields for the MERGE. The records in each input data set to be merged must already be in
sorted order as specified by the control field in the supplied DFSORT MERGE statement. Additional DFSORT
control statements and options can be used to merge a subset of the input records (INCLUDE or OMIT statement;
OUTFIL INCLUDE, OMIT, SAVE, STARTREC, ENDREC, ACCEPT, SAMPLE, SPLIT, SPLITBY and SPLIT1R
operands), reformat records for output (INREC, OUTREC and OUTFIL statements), and so on. See z/OS DFSORT
Application Programming Guide for complete details of DFSORT control statements and options.

 Example
MERGE FROM(INPUT1,INPUT2,INPUT3,INPUT4,INPUT5)-
 FROM(INPUT6,INPUT7) USING(CTL1)

CTL1CNTL might contain the following:

//CTL1CNTL DD *
 MERGE FIELDS=(52,8,UFF,D)
 OUTFIL FNAMES=OUT1,INCLUDE=(15,3,SS,EQ,C'D21,D33')
 OUTFIL FNAMES=OUT2,SAVE

 Required Operands
� FROM - the ddnames of 1 to 50 input data sets. You must supply DD statements for the ddnames you specify.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL. You must supply a DFSORT MERGE statement in xxxxCNTL.

 Optional Operands
� TO - the ddnames of 1 to 10 output data sets. You must supply DD statements for the ddnames you specify.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� LOCALE - overrides the installation default for locale processing.

� SERIAL - causes OUTFIL processing not to be used for multiple outdd data sets. SERIAL is not recom-
mended because it imposes data set restrictions and can degrade performance.

 Note
� If you use OUTFIL statements in xxxxCNTL to specify your output data sets, you do not need to specify TO.

Example of TOOLMSG Output for MERGE
* MERGE - merges one or more data sets to one or more
* output data sets.
* Example: merge the IN01, IN02 and IN03 data sets to
* the MRGOUT output data set using the DFSORT control
* statements in the MRG1CNTL data set.
MERGE FROM(IN01,IN02,IN03) TO(MRGOUT) USING(MRG1)

ICE606I 0 DFSORT CALL 0005 FOR MERGE FROM MERGEIN TO MRGOUT
USING MRG1CNTL COMPLETED

ICE602I 0 OPERATION RETURN CODE: 00

MODE Operator Details

 Syntax
MODE STOP
 CONTINUE
 SCAN

 Function

Specifies one of three modes to control error checking and actions after error detection. A MODE operator affects
the processing of the group of operators which follow it, up to the next MODE operator (if any). Dependent
operators (those for which a failure of one should stop execution of the rest) can be grouped with MODE STOP.
Independent operators (those for which a failure of one should not affect execution of the rest) can be grouped with
MODE CONTINUE. Operators to be checked for errors only can be grouped with MODE SCAN.

 Examples
MODE CONTINUE
 <independent operator group>

MODE STOP
 <dependent operator group>

 Required Operand

STOP, CONTINUE or SCAN.

� STOP - If an operation fails, stops processing the remaining operators in the group, but continues to check for
errors in ICETOOL statements. STOP mode is the default set at the beginning of an ICETOOL run.

� CONTINUE - If an operation fails, continues processing the remaining operators in the group.

� SCAN - Checks for errors in ICETOOL statements without processing the operators. Set automatically if an
error is detected while in STOP mode.

 Note
� The return codes for one group of operators does not affect the other groups of operators.

Example of TOOLMSG Output for MODE

* MODE - specifies the error checking and actions after error
* detection to be performed for a group of operators.
* Example: continue processing operators whether or not an
* error is detected.

 MODE CONTINUE
ICE630I 1 MODE IN EFFECT: CONTINUE
ICE602I 1 OPERATION RETURN CODE: 00

OCCUR Operator Details

 Syntax
OCCUR FROM(indd) LIST(listdd) ON(p,m,f) ...
OCCURS ON(p,m,f,formatting)
 ON(p,m,HEX)
 ON(VLEN)
 ON(VLEN,formatting)
 ON(VALCNT)
 ON(VALCNT,formatting)

TITLE('string1') ... TLEFT TFIRST
 TITLE('string1','string2')
 TITLE('string1','string2','string3')

 PAGE DATE TIME NOCC
 DATE(abcd) TIME(abc)
 DATENS(abc) TIMENS(ab)
 YDDD(abc)
 YDDDNS(ab)

HEADER('string1') ... LINES(n) BLANK
 HEADER('string1','string2') PLUS
 HEADER('string1','string2','string3')
 HEADER(NONE)
 NOHEADER

 ALLDUPS VSAMTYPE(x) WIDTH(n)
 NODUPS
 HIGHER(x)
 LOWER(y)
 EQUAL(v)

INDENT(n) BETWEEN(n) TBETWEEN(n) UZERO LISTSDB
 LISTNOSDB

 Function

Prints, in the listdd data set, each unique value for specified numeric and/or character fields and how many times it
occurs. From 1 to 10 ON fields can be specified as long as the resulting list data set line length does not exceed
the limit specified by the WIDTH operand or 8192 bytes. All ON fields specified are used to determine whether a
record contains a unique value. A single list data set record is printed for each unique value with the fields printed
in columns in the same order in which they are specified.

Simple or tailored reports can be produced.

� Up to three title lines can appear at the top of each page or only at the top of the first page. The first title line
can consist of a title string, the page number, the date and/or the time, in any order specified. The second title
line consists of a specified title string. The third title line consists of a specified title string. The title strings
can be centered or left-justified with respect to each other.

� Carriage control characters can be printed or suppressed.

� One, two or three line headings can be printed to identify each specified field.

� Column widths are adjusted automatically according to the operands specified and the length of the fields.

� Alternate formats for date and time can be specified.

� Numeric fields, consisting of a sign and up to 31 digits in various formats, can be edited with separators,
decimal points, patterns, and leading, trailing and floating signs.

The ON values printed can be limited to only duplicates, non-duplicates or those that occur less than, equal to or
more than n times.

 Examples
OCCUR FROM(SOURCE) LIST(VOLSERS) ON(40,6,CH) ON(VALCNT)

OCCUR FROM(FAILURES) LIST(CHECKIT) LISTNOSDB -
DATE(YMD.) TITLE('Possible System Intruders') TIME(12.) -
TITLE('based on logon failures') TFIRST -
HEADER(' Userid ') ON(23,8,CH) -
HEADER(' Logon Failures ') ON(VALCNT) -

 HIGHER(4) BLANK

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� LIST - the ddname of the list data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation. From 1 to 10 ON fields can be specified.

– (p,m,f) gives the position, length and format of a numeric or character field. A field must not extend
beyond position 32752 or the end of the record.

– (p,m,f,formatting) gives the position, length and format of a numeric or character field and specifies how
the data for this field is to be formatted for printing. A field must not extend beyond position 32752 or the
end of the record.

The following formatting items can be used (see DFSORT Application Programming Guide for complete
details):

— Mask - an edit mask to be applied to the numeric data. Thirty-nine pre-defined edit masks are avail-
able, encompassing many of the numeric notations throughout the world with respect to separators,
decimal point, decimal places, signs and so on. See "Appendix A" for complete descriptions and
examples of all thirty-nine masks. The attributes of each group of masks is shown below.

— E'pattern' - an edit pattern (1 to 44 characters) to be applied to the numeric data. Each 9 in the pattern
(up to 31) is replaced by a corresponding digit (0-9) from the numeric value. Other characters in the
pattern appear as specified. E'pattern' can be used for formatting unsigned numeric data such as tele-
phone numbers, dates, and so on.

— L'string' - a leading string to appear at the beginning of the character or numeric data column. Each
string can be 1 to 10 characters.

— F'string' - a floating string to appear to the left of numeric data. Each string can be 1 to 10 characters.

— T'string' - a trailing string to appear at the end of the character or numeric data column. Each string
can be 1 to 10 characters.

— LZ - used with an edit mask to print leading zeros for the numeric field (overriding the default of
suppressing leading zeros).

— Ndd - use dd digits for the numeric field where dd is greater than the calculated number of digits. dd
must be a two-digit number from 01 to 31.

— Udd - use dd digits for the numeric field where dd is greater than or less than the calculated number
of digits. dd must be a two-digit number from 01 to 31.

– (p,m,HEX) gives the position and length of a character field to be printed in hexadecimal format. A field
must not extend beyond position 32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record. The following formatting items can be used with VLEN: mask, E'pattern', L'string', F'string',
T'string', LZ, Ndd and Udd.

– VALCNT causes the count of occurrences for each value to be printed. The following formatting items
can be used with VALCNT: mask, E'pattern', L'string', F'string', T'string', LZ, Ndd and Udd.

– A description of each type of field is given below.

 
Masks

 
Separators

Decimal
Places

Positive
Sign

Negative
Sign

A0 No 0 blank -

A1-A5 Yes 0 blank -

B1-B6 Yes 1 blank -

C1-C6 Yes 2 blank -

D1-D6 Yes 3 blank -

E1-E4 Yes 0 blank ( )

F1-F5 Yes 2 blank ( )

G1-G6 Yes 4 blank -

Format Code Length Description

BI 1-8 bytes Unsigned binary

FI 1-8 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CH 1-4000 bytes Character

 Optional Operands
� TITLE, PAGE, DATE, DATENS, YDDD, YDDDNS, TIME and TIMENS - the elements to appear in the title

lines at the top of each page. Only specified elements appear and in the order given. Each title string can
consist of up to three individual strings. The page, date and time appear in the first title line with the first title
string. The second title string appears in the second title line. The third title string appears in the third title
line. A string can be 1 to 50 characters; the maximum length for each title string can be 50 characters.
Alternate formats for date and time can be specified.

� TLEFT - title strings are left-justified with respect to each other (overriding the default of centering title strings
with respect to each other).

� TFIRST - the title lines are only printed on the first page (overriding the default of printing the title lines on
every page).

Format Code Length Description

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

DT1 4 bytes SMF date as Z'yyyymmdd'

DT2 4 bytes SMF date as Z'yyyymm'

DT3 4 bytes SMF date as Z'yyyyddd'

DC1 8 bytes TOD date as Z'yyyymmdd'

DC2 8 bytes TOD date as Z'yyyymm'

DC3 8 bytes TOD date as Z'yyyyddd'

DE1 8 bytes ETOD date as Z'yyyymmdd'

DE2 8 bytes ETOD date as Z'yyyymm'

DE3 8 bytes ETOD date as Z'yyyyddd'

TM1 4 bytes SMF time as Z'hhmmss'

TM2 4 bytes SMF time as Z'hhmm'

TM3 4 bytes SMF time as Z'hh'

TM4 4 bytes SMF time as Z'hhmmssxx'

TC1 8 bytes TOD time as Z'hhmmss'

TC2 8 bytes TOD time as Z'hhmm'

TC3 8 bytes TOD time as Z'hh'

TC4 8 bytes TOD time as Z'hhmmssxx'

TE1 8 bytes ETOD time as Z'hhmmss'

TE2 8 bytes ETOD time as Z'hhmm'

TE3 8 bytes ETOD time as Z'hh'

TE4 8 bytes ETOD time as Z'hhmmssxx'

HEX 1-2000 bytes Character printed as hexadecimal

VLEN n/a Record length for VLR (1,2,BI)

VALCNT n/a Value count

� NOCC - suppresses carriage control characters (overriding the default of printing carriage control characters).

� HEADER - a one, two or three line heading to be printed for the corresponding ON field (overriding the
"standard" one line heading of "(p,m,f)"). A heading string can be 1 to 50 characters. NONE can be used to
suppress the heading for the corresponding ON field.

� NOHEADER - suppresses the heading line.

� LINES - the number of lines per page (overriding the default of 58). n can be 10 to 999.

� BLANK and PLUS - causes the column widths to be dynamically adjusted as needed (overriding the standard
fixed column widths) and suppresses leading zeros for numeric fields. BLANK causes a blank, rather than a +
to be used for the positive sign.

� ALLDUPS, NODUPS, HIGHER, LOWER, EQUAL - limits the values to be printed to those whose occurences
meet the given criteria. x, y, and v must be specified as n or +n where n can be 1 to 15 decimal digits.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� WIDTH - the line length and LRECL for the list data set. n can be 121 to 8192 (or 121 to 8191 if NOCC is
specified). If the WIDTH specified is less than the calculated line length, ICETOOL terminates the operation.
If WIDTH is not specified, the calculated line length (subject to a minimum of 121, or 120 if NOCC is speci-
fied), is used as the line length and LRECL for the list data set.

� LONGLINE - if WIDTH(n) is not specified, allows the calculated line length to be up to 8192 bytes (over-
riding the default of 2048 bytes).

� INDENT - the number of blanks used to indent the report (overriding the default of 0). n can be 0 to 50.

� BETWEEN - the number of blanks used between the columns of data (overriding the default of 3). n can be 0
to 50.

� TBETWEEN - the number of blanks used between title elements (overriding the default of 8). n can be 0 to
50.

� UZERO - causes -0 to be treated as unsigned, that is, as +0.

� LISTSDB - uses SDB for the list data set (overrides installation value SDBMSG=NO).

� LISTNOSDB - does not use SDB for the list data set (overrides installation value SDBMSG=YES).

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

� Symbols for character constants can be used instead of 'string' in the TITLE and HEADER operands.

� Symbols for decimal constants can be used instead of n and +n in the HIGHER, LOWER and EQUAL oper-
ands.

 Notes
� The control character occupies the first byte of each record in the list data set unless NOCC is specified.

� By default, the first column starts in the second byte of each record in the list data set (or in the first byte if
NOCC is specified). INDENT(n) can be used to override this default.

� By default, three blanks appear between columns in the list data set. BETWEEN(n) can be used to override
this default.

� By default, eight blanks appear between title elements in the list data set. TBETWEEN(n) can be used to
override this default.

� RECFM=FBA is used for the list data set if NOCC is not specified. RECFM=FB is used for the list data set if
NOCC is specified.

� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort performed for the
OCCUR operation.

� Tape work data sets cannot be used with ICETOOL.

� Specifying LONGLINE can enable you to include more fields in your report, up to a maximum of 10.

Example of TOOLMSG Output for OCCUR
* OCCUR - prints each unique value for specified numeric and
* character fields and how many times it occurs in a separate
* list data set. Simple or tailored reports can be produced.
* The values printed can be limited to those for which the
* value meets specified criteria (e.g. only duplicate values).
* Example: print a report showing values from the IN1 data set
* ON field, with the number of times each value occurs, in the
* LIST1 data set.
OCCUR FROM(IN1) LIST(LIST1) TITLE('Dep''t Counts') -

 ON(35,3,CH) ON(VALCNT)
ICE627I 0 DFSORT CALL 0006 FOR SORT FROM IN1 TO E35 EXIT COMPLETED
ICE603I 0 INFORMATION PRINTED IN LIST1 DATA SET
ICE628I 0 RECORD COUNT: 000000000000072
ICE638I 0 NUMBER OF RECORDS RESULTING FROM CRITERIA: 000000000000005
ICE602I 0 OPERATION RETURN CODE: 00

Example of List Output for OCCUR

Here's an example of the output that would appear in LIST1:

Dep't Counts

(35,3,CH) VALUE COUNT
D54 000000000000012
J69 000000000000009
J82 000000000000020
L92 000000000000017
M27 000000000000014

Example of HIGHER with OCCUR
OCCUR FROM(CALLERS) LIST(CALLRPT) -
DATENS(4MD) TITLE('Large Call Volume') -
INDENT(5) BETWEEN(10) -
HEADER('Phone Number') ON(7,10,ZD,E'(999)-999-9999')-

 HEADER('Calls') ON(VALCNT,A1,N06)-
 HIGHER(1000)

Example of HIGHER with OCCUR Output
20050228 Large Call Volume

 Phone Number Calls
 -------------- --------
 (037)-325-1807 3,521
 (216)-721-5530 2,100
 (856)-003-3008 12,163

RANGE Operator Details

 Syntax
RANGE FROM(indd) ON(p,m,f) HIGHER(x) VSAMTYPE(x)
 ON(VLEN) LOWER(y)
 EQUAL(v)
 NOTEQUAL(w)

 Function

Prints a message in TOOLMSG containing the count of values in a specified range for a specific numeric field.
The range can be specified as higher than x, lower than y, higher than x and lower than y, equal to v, or not equal
to w, where x, y, v, and w are signed or unsigned decimal values.

 Examples
RANGE FROM(DATA1) ON(VLEN) HIGHER(52)

RANGE FROM(DATA2) ON(31,18,ZD) EQUAL(-123456789012345678)

RANGE FROM(DATA2) ON(25,3,PD) HIGHER(-20) LOWER(+15)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation.

– (p,m,f) gives the position, length and format of a numeric field. A field must not extend beyond position
32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record.

– A description of each type of field is given below.

� HIGHER, LOWER, EQUAL, NOTEQUAL - defines the range for the values to be counted. HIGHER and
LOWER may be used together or separately. EQUAL and NOTEQUAL must be used separately. x, y, v, and
w must be specified as n, +n, or -n where n can be 1 to 31 digits.

Format Code Length Description

BI 1-8 bytes Unsigned binary

FI 1-8 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

VLEN n/a Record length for VLR (1,2,BI)

 Optional Operand
� VSAMTYPE - the record format for a VSAM input data set (F or V).

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

� Symbols for decimal constants can be used instead of n, +n and -n in the HIGHER, LOWER, EQUAL and
NOTEQUAL operands.

 Note
� If the range is specified as HIGHER(x) LOWER(y), it must be a valid range. For example, HIGHER(5)

LOWER(6) is not a valid range since there is no integer value that satisfies the criteria.

Example of TOOLMSG Output for RANGE
* RANGE - prints a message in TOOLMSG containing the count
* of values in a specified range for a specified numeric
* field in a data set.
* Example: print a count of the values in the IN2 data set
* ON field that are higher than -50, but lower than +100.
RANGE FROM(IN2) ON(20,2,PD) HIGHER(-50) LOWER(100)

ICE627I 0 DFSORT CALL 0007 FOR COPY FROM IN2 TO E35 EXIT COMPLETED
ICE628I 0 RECORD COUNT: 000000000000008
ICE631I 0 NUMBER OF VALUES IN RANGE FOR (20,2,PD) : 000000000000004
ICE602I 0 OPERATION RETURN CODE: 00

RESIZE Operator Details

 Syntax
RESIZE FROM(indd) TO(outdd) TOLEN(n) USING(xxxx)

 Function

Produces fixed length output records of a specified length from fixed length input records of a different length. If
the input length is smaller than the specified length, multiple smaller input records are combined into one larger
output record. If the input length is larger than the specified length, each larger input record is broken up into
multiple smaller output records.

The DFSORT control statements in xxxxCNTL are used if USING(xxxx) is specified.

 Examples
RESIZE FROM(LARGE) TO(SMALL) TOLEN(25)

RESIZE FROM(OLD) TO(NEW) TOLEN(120) USING(CTL1)

CTL1CNTL might contain the following:

OMIT COND=(1,30,CH,EQ,C' ')
 SORT FIELDS=(1,30,CH,A)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� TO - the ddname of the output data set. You must supply a DD statement for the ddname you specify.

� TOLEN - the record length you want ICETOOL to use for the resized output records. n can be 1 to 32760.

 Optional Operand
� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You

must supply a DD statement for xxxxCNTL if you specify USING(xxxx). See z/OS DFSORT Application
Programming Guide for details of the DFSORT control statements you can use with RESIZE.

 Notes
� The value of n you specify for TOLEN must not be equal to the input record length.

� If you specify a DFSORT SORT statement in xxxxCNTL, the DFSORT DYNALLOC option is used to ensure
that work space is available for the sort performed for the RESIZE operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for RESIZE

* RESIZE - creates larger records from smaller records, or
* smaller records from larger records.
* Example: create 40 byte output records from 10 byte input

 * records.
RESIZE FROM(L10) TO(L40) TOLEN(40)

ICE627I 0 DFSORT CALL 0008 FOR COPY FROM L10 TO L40 COMPLETED
ICE659I 0 RECORDS RESIZED FROM 00010 BYTES TO 00040 BYTES
ICE602I 0 OPERATION RETURN CODE: 00

SELECT Operator Details

 Syntax
SELECT FROM(indd) TO(outdd) ON(p,m,f) ...
 DISCARD(savedd) ON(VLEN)

ALLDUPS VSAMTYPE(x) UZERO USING(xxxx)
 NODUPS
 HIGHER(x)
 LOWER(y)
 EQUAL(v)
 FIRST
 FIRST(u)
 LAST
 FIRSTDUP
 FIRSTDUP(w)
 LASTDUP

 Function

Selects records from the indd data set for inclusion in the outdd data set based on meeting criteria for the number of
times specified numeric and/or character field values occur. From 1 to 10 ON fields can be specified. All ON
fields are used to determine the value count (that is, the number of times the ON values occur) to be matched
against the criteria.

The records selected can be limited to those with duplicate values, non-duplicate values, values that occur less than,
equal to or more than n times, or the first, first n, or last record with each unique or duplicate value.

DISCARD(savedd) can be used to save the records which do not meet the criteria (that is, the discarded records), in
the savedd data set. DISCARD(savedd) may be used with or without TO(outdd).

The DFSORT control statements in xxxxCNTL are used if USING(xxxx) is specified.

 Examples
SELECT FROM(INPUT) TO(DUPS) ON(11,8,CH) ON(30,44,CH) ALLDUPS

SELECT FROM(IN) TO(OUT1) ON(29,5,FS) HIGHER(3)

SELECT FROM(DATAX) TO(NEWEST) ON(135,20,ZD) LASTDUP USING(CTL1)

SELECT FROM(BOOKS) TO(PUBLISHR) ON(29,10,UFF) FIRST(3) -
 DISCARD(SAVEREST)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� TO - the ddname of the output data set for the selected records. You must supply a DD statement for the
ddname you specify.

� DISCARD - the ddname of the output data set for the records which are not selected. You must supply a DD
statement for the ddname you specify. TO(outdd) and DISCARD(savedd) may be used together or separately.

� ON - a field to be used for this operation. From 1 to 10 ON fields can be specified.

– (p,m,f) gives the position, length and format of a numeric or character field. A field must not extend
beyond position 32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record.

– A description of each type of field is given below.

� ALLDUPS, NODUPS, HIGHER, LOWER, EQUAL, FIRST, LAST, FIRSTDUP, LASTDUP - defines the cri-
teria against which the value counts are to be matched. x, y, v, u and w must be specified as n or +n where n
can 0 to 99.

 Optional Operands
� VSAMTYPE - the record format for a VSAM input data set (F or V).

� UZERO - causes -0 to be treated as unsigned, that is, as +0.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL if you specify USING(xxxx).

If used, the xxxxCNTL data set should only contain one or more of these statements:

– INCLUDE or OMIT - deleted records are not processed by SELECT.

– INREC - the reformatted records are processed by SELECT.

– SORT - additional sort fields can be specified after p,m,f,A fields corresponding to ON(p,m,f) fields. The
additional sort fields are used for sorting, but not for selecting.

– OPTION - for optional parameters like MAINSIZE.

– OUTFIL - the selected records are processed by OUTFIL. With TO(outdd), use a corresponding OUTFIL
statement with FNAMES=outdd. With DISCARD(savedd), use a corresponding OUTFIL statement with

Format Code Length Description

BI 1-1500 bytes Unsigned binary

FI 1-256 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CH 1-4000 bytes Character

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

VLEN n/a Record length for VLR (1,2,BI)

FNAMES=savedd. With TO(outdd) and DISCARD(savedd), use corresponding OUTFIL statements with
FNAMES=outdd and FNAMES=savedd.

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

� Symbols for decimal constants can be used instead of n and +n in the HIGHER, LOWER, EQUAL, FIRST and
FIRSTDUP operands.

 Notes
� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort performed for the

SELECT operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for SELECT
* SELECT - selects records from a data set for inclusion
* in an output data set based on meeting criteria for the
* number of times specified numeric or character field
* values occur (e.g. only duplicate values).
* Records that are not selected can be saved in a separate
* output data set.
* Example: selects records from the IN1 data set, for the
* SEL1 data set, whose ON field occurs only once (i.e, only
* records with no duplicate ON field values).
SELECT FROM(IN1) TO(SEL1) ON(30,2,PD) NODUPS

ICE627I 0 DFSORT CALL 0009 FOR SORT FROM IN1 TO SEL1 COMPLETED
ICE628I 0 RECORD COUNT: 000000000000072
ICE638I 0 NUMBER OF RECORDS RESULTING FROM CRITERIA: 000000000000013
ICE602I 0 OPERATION RETURN CODE: 00

SORT Operator Details

 Syntax
SORT FROM(indd) USING(xxxx) TO(outdd,...) VSAMTYPE(x)
 JKFROM

 LOCALE(name) SERIAL
 LOCALE(CURRENT)
 LOCALE(NONE)

 Function

Sorts the indd data set to the outdd data sets (up to 10) using the the DFSORT control statements in xxxxCNTL.
You must supply a DFSORT SORT statement in xxxxCNTL to indicate the control fields for the sort. Additional
DFSORT control statements and options can be used to sort a subset of the input records (INCLUDE or OMIT
statement; SKIPREC and STOPAFT options; OUTFIL INCLUDE, OMIT, SAVE, STARTREC, ENDREC,
ACCEPT, SAMPLE, SPLIT, SPLITBY and SPLIT1R operands), reformat records for output (INREC, OUTREC
and OUTFIL statements), and so on. See z/OS DFSORT Application Programming Guide for complete details of
DFSORT control statements and options.

 Examples
SORT FROM(A) TO(B,C,D) USING(DEPT)

SORT FROM(VSAMIN) TO(VSAMOUT1) USING(VSAM) -
 VSAMTYPE(F) LOCALE(DA_DK)

SORT FROM(MASTER) USING(MULT)

DEPTCNTL might contain the following:

//DEPTCNTL DD *
 SORT FIELDS=(15,3,A,27,4,D),FORMAT=PD
 INREC OVERLAY=(22:5,8,SQZ=(SHIFT=LEFT))
/*

VSAMCNTL might contain the following:

//VSAMCNTL DD *
 SORT FIELDS=(27,12,CH,A)
/*

MULTCNTL might contain the following:

//MULTCNTL DD *
 SORT FIELDS=(18,3,ZD,A,43,2,BI,D)
 OUTFIL FNAMES=DEPT1,INCLUDE=(5,3,CH,EQ,C'D01')
 OUTFIL FNAMES=DEPT2,INCLUDE=(5,3,CH,EQ,C'D02')
 OUTFIL FNAMES=DEPT3,INCLUDE=(5,3,CH,EQ,C'D03')
 OUTFIL FNAMES=REST,SAVE
/*

Note that DFSORT's OUTFIL INCLUDE and SAVE features are used here to create four different subset data sets
using a single pass over the input data set. SAVE allows you to easily include the records that are not included in

any other OUTFIL data set. Complete details on this feature of OUTFIL as well as its many other features can be
found in DFSORT Application Programming Guide.

 Required Operands
� FROM or JKFROM

– FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

– JKFROM - used for a JOINKEYS application. You must provide a USING(xxxx) operand. In
xxxxCNTL, you must provide a JOINKEYS statement with F1=ddname1 for the F1 file and a JOINKEYS
statement with F2=ddname2 for the F2 file, as well as JOIN and REFORMAT statements as needed.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL. You must supply a DFSORT SORT statement in xxxxCNTL.

 Optional Operands
� TO - the ddnames of 1 to 10 output data sets. You must supply DD statements for the ddnames you specify.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� LOCALE - overrides the installation default for locale processing.

� SERIAL - causes OUTFIL processing not to be used for multiple outdd data sets. SERIAL is not recom-
mended because it imposes data set restrictions and can degrade performance.

 Notes
� If you use OUTFIL statements in xxxxCNTL to specify your output data sets, you do not need to specify TO.

� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for SORT
* SORT - sorts a data set to one or more output data sets.
* Multiple output is handled using a single pass over
* the input.
* Example: sort the IN1 data set to the OUT1 data set using
* the DFSORT control statements in the CTL1CNTL data set.
SORT FROM(IN1) TO(OUT1) USING(CTL1)

ICE606I 0 DFSORT CALL 0010 FOR SORT FROM IN1 TO OUT1
USING CTL1CNTL COMPLETED

ICE602I 0 OPERATION RETURN CODE: 00

SPLICE Operator Details

 Syntax
SPLICE FROM(indd) TO(outdd) ON(p,m,f) ... WITH(p,m) ...

WITHALL KEEPNODUPS KEEPBASE VSAMTYPE(x) UZERO USING(xxxx)
 WITHANY
 WITHEACH

 VLENMAX
 VLENOVLY

 Function

Splices together specified fields from records with matching numeric or character field values (that is, duplicate
values), but different information. This makes it possible to join fields from different types of input records to
create an output record with information from two or more records. From 1 to 10 ON fields can be specified. All
ON fields are used to determine if records match.

Typically, you will want to reformat the records from two or more data sets to a temporary MOD data set, and use
that temporary MOD data set as input to the SPLICE operator.

By default, the first duplicate is spliced with all of the WITH fields from the last duplicate. From 1 to 50 WITH
fields can be specified. WITHALL can be used to splice the first duplicate with all of the WITH fields from the
second and subsequent duplicates. WITHANY can be used to splice the first duplicate with nonblank WITH fields
from the second and subsequent duplicates. WITHEACH can be used to splice the first duplicate with one WITH
field from the second and subsequent duplicates in turn.

By default, non-matching records are not kept. KEEPNODUPS can be used to keep non-matching records.

By default, the base record is not kept. KEEPBASE can be used to keep the base record.

By default, for variable-length records, the length of the base record is used as the length of the spliced record.
VLENMAX can be used to set the length of the spliced record to the maximum length of the base record or overlay
record. VLENOVLY can be used to set the length of the spliced record to the length of the overlay record.

The DFSORT control statements in xxxxCNTL are used if USING(xxxx) is specified.

 Examples
SPLICE FROM(T1) TO(OUT1) ON(11,8,CH) ON(30,44,CH) WITH(1,100)

SPLICE FROM(T2) TO(OUT2) ON(11,8,CH) -
WITHALL WITH(51,50) WITH(101,75) -

 KEEPNODUPS USING(CTL1)

SPLICE FROM(ROWS) TO(OUTPUT) ON(15,5,ZD) -
WITHANY WITH(21,20) WITH(41,20) WITH(61,20)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� TO - the ddname of the output data set for the spliced records. You must supply a DD statement for the
ddname you specify.

� ON - a field to be used for this operation. From 1 to 10 ON fields can be specified.

– (p,m,f) gives the position, length and format of a numeric or character field. A field must not extend
beyond position 32752 or the end of the record.

– A description of each type of field is given below.

� WITH(p,m) - a field to be spliced from the second or subsequent duplicate record to the first duplicate record.
From 1 to 50 WITH fields can be specified. (p,m) gives the position and length of the field.

 Optional Operands
� WITHALL - the first duplicate is spliced with all of the specified WITH fields from the second duplicate, and

then from each subsequent duplicate in turn (overriding the default of splicing the first duplicate with all of the
specified WITH fields from the last duplicate).

� WITHANY - the first duplicate is spliced with all of the specified nonblank WITH fields from each subsequent
duplicate (overriding the default of splicing the first duplicate with all of the specified WITH fields from the
last duplicate).

� WITHEACH - the first duplicate is spliced with one specified WITH field from each subsequent duplicate
(overriding the default of splicing the first duplicate with all of the specified WITH fields from the last dupli-
cate).

� KEEPNODUPS - non-duplicates are kept (unchanged) along with the spliced records (overriding the default of
deleting non-duplicates).

� KEEPBASE - the base records are kept (unchanged) along with the spliced records (overriding the default of
deleting the base records).

� VSAMTYPE - the record format for a VSAM input data set (F or V).

� UZERO - causes -0 to be treated as unsigned, that is, as +0.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL if you specify USING(xxxx).

If used, the xxxxCNTL data set should only contain one or more of these statements:

– INCLUDE or OMIT - deleted records are not processed by SPLICE.

Format Code Length Description

BI 1-1500 bytes Unsigned binary

FI 1-256 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CH 1-4000 bytes Character

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

– INREC - the reformatted records are processed by SPLICE.

– OPTION - for optional parameters like MAINSIZE.

– OUTFIL - the selected records (spliced records, and unspliced records if KEEPNODUPS or KEEPBASE is
specified) are processed by OUTFIL. With TO(outdd), use a corresponding OUTFIL statement with
FNAMES=outdd.

� VLENMAX - for variable-length records, the length of the spliced record is set to the maximum length of the
base record or overlay record (overriding the default of setting the length of the spliced record to the length of
the base record).

� VLENOVLY - for variable-length records, the length of the spliced record is set to the length of the overlay
record (overriding the default of setting the length of the spliced record to the length of the base record).

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand and instead of p,m in the WITH

operand.

 Notes
� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort performed for the

SPLICE operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for SPLICE
* SPLICE - splices together specified fields from records
* that have the same specified numeric or character field
* values (i.e., duplicate values), but different
* information. Specified fields from two or more records
* can be combined to create an output record. The fields
* to be spliced can originate from records in different
* data sets, so various "join" and "match" operations can
* be performed.
* Example: for ON fields that occur more than once (i.e.,
* duplicate ON field values) in the CONCT data set, splices
* the WITH field from the last duplicate record into the
* first duplicate record.
SPLICE FROM(CONCT) TO(COMBINE) ON(11,5,CH) WITH(41,20)

ICE627I 0 DFSORT CALL 0011 FOR SORT FROM CONCT TO COMBINE COMPLETED
ICE628I 0 RECORD COUNT: 000000000000019
ICE638I 0 NUMBER OF RECORDS RESULTING FROM CRITERIA: 000000000000007
ICE602I 0 OPERATION RETURN CODE: 00

STATS Operator Details

 Syntax
STATS FROM(indd) ON(p,m,f) VSAMTYPE(x) LMSG
 ON(VLEN)

 Function

Prints messages in TOOLMSG containing the minimum, maximum, average and total for up to 10 specified
numeric fields. The average is calculated by dividing the total by the record count and rounding down to the
nearest integer.

 Examples
STATS FROM(DATA1) ON(VLEN) ON(15,4,ZD)

STATS FROM(VSAMIN) ON(5,8,BI) ON(20,2,PD) ON(12,6,FS) ON(47,3,FI) -
 VSAMTYPE(F)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation.

– (p,m,f) gives the position, length and format of a numeric field. A field must not extend beyond position
32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record.

– A description of each type of field is given below.

 Optional Operand
� VSAMTYPE - the record format for a VSAM input data set (F or V).

� LMSG - the minimum, maximum, average and total for all numeric fields are printed using messages that
display 31 digits (overriding the default of printing messages that display 15 digits when possible).

Format Code Length Description

BI 1-8 bytes Unsigned binary

FI 1-8 bytes Signed fixed-point

PD 1-16 bytes Signed packed decimal

ZD 1-31 bytes Signed zoned decimal

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

VLEN n/a Record length for VLR (1,2,BI)

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

 Note
� If the total for a field overflows, ICETOOL continues processing, but prints asterisks for the average and total

for that field.

Example of TOOLMSG Output for STATS
* STATS - prints messages in TOOLMSG containing the minimum,
* maximum, average, and total for specified numeric fields in
* a data set.
* Example: print the minimum, maximum, average and total
* values for the three VLRIN data set ON fields.
* For variable-length records, ON(VLEN) gives statistics
* about the length of the records.
STATS FROM(VLRIN) ON(VLEN) ON(12,2,ZD) ON(18,5,FS)

ICE627I 0 DFSORT CALL 0012 FOR COPY FROM VLRIN TO E35 EXIT COMPLETED
ICE628I 0 RECORD COUNT: 000000000000017
ICE607I 0 STATISTICS FOR (VLEN) :
ICE608I 0 MINIMUM: +000000000000058, MAXIMUM: +000000000000079
ICE609I 0 AVERAGE: +000000000000068, TOTAL : +000000000001171
ICE607I 0 STATISTICS FOR (12,2,ZD) :
ICE608I 0 MINIMUM: -000000000000064, MAXIMUM: +000000000000082
ICE609I 0 AVERAGE: +000000000000010, TOTAL : +000000000000177
ICE607I 0 STATISTICS FOR (18,5,FS) :
ICE608I 0 MINIMUM: -000000000003892, MAXIMUM: +000000000018723
ICE609I 0 AVERAGE: +000000000001127, TOTAL : +000000000019168
ICE602I 0 OPERATION RETURN CODE: 00

SUBSET Operator Details

 Syntax
SUBSET FROM(indd) TO(outdd) KEEP INPUT

DISCARD(savedd) REMOVE OUTPUT

HEADER TRAILER RRN(q) ...
 FIRST LAST RRN(q,r)

HEADER(u) TRAILER(v) RRN(q,*)
 FIRST(u) LAST(v)

 USING(xxxx) VSAMTYPE(x)

 Function

Selects input or output records from the indd data set for inclusion in the outdd data set based on meeting criteria
for keeping or removing the first n records, specific relative record numbers, and/or the last n records. From 1 to
300 relative records or ranges of relative records can be specified.

DISCARD(savedd) can be used to save the records which do not meet the criteria (that is, the discarded records).
DISCARD(savedd) may be used with or without TO(outdd).

The DFSORT control statements in xxxxCNTL are used if USING(xxxx) is specified.

 Examples
SUBSET FROM(INPUT) TO(OUTPUT) REMOVE INPUT HEADER TRAILER

SUBSET FROM(FILE1) TO(FILE2) REMOVE INPUT -
RRN(8) RRN(21,25) RRN(5001,*)

SUBSET FROM(IN1) TO(OUT1) REMOVE OUTPUT LAST(5) USING(SRT1)

SRT1CNTL might contain the following:

 SORT FIELDS=(11,5,CH,A,21,6,SFF,D)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� TO - the ddname of the output data set for the selected records (the records that are kept or not removed
according to the specified criteria). You must supply a DD statement for the ddname you specify.

� DISCARD - the ddname of the output data set for the records which are not selected. You must supply a DD
statement for the ddname you specify. TO(outdd) and DISCARD(savedd) may be used together or separately.

� KEEP, REMOVE - indicates whether the records that meet the criteria are kept or removed.

� INPUT, OUTPUT - indicates whether the criteria are to be applied to the input records or output records.

 Optional Operands
� HEADER, FIRST, TRAILER, LAST, RRN - specifies the header records (first u records), trailer records (last v

records) and/or relative records (relative record q, or relative records q to r) to be kept or removed. RRN(q,*)
can be used to specify relative records q through the last record. From 1 to 300 RRN operands can be speci-
fied. u, v, q and r must be specified as n or +n where n can be 1 to 999999999999999.

� USING - the first 4 characters of the ddname (xxxxCNTL) for the DFSORT control statement data set. You
must supply a DD statement for xxxxCNTL if you specify USING(xxxx). See z/OS DFSORT Application
Programming Guide for details of the DFSORT control statements you can use with SUBSET.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

 Symbols
� Symbols for decimal constants can be used instead of n and +n in the HEADER, FIRST, TRAILER, LAST and

RRN operands.

 Notes
� If you specify a DFSORT SORT statement in xxxxCNTL, the DFSORT DYNALLOC option is used to ensure

that work space is available for the sort performed for the SUBSET operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for SUBSET
* SUBSET - selects records from a data set based on keeping or
* removing header records (the first n records), relative records,
* or trailer records (the last n records).
* Records that are not selected can be saved in a separate
* output data set.
* Example: Copies records 5, 21 and 31-33 from the
* IN1 data set to the SUB1 data set.
SUBSET FROM(IN1) TO(SUB1) INPUT KEEP -
RRN(5) RRN(21) RRN(31,33)

ICE627I 0 DFSORT CALL 0013 FOR COPY FROM IN1 TO SUB1 COMPLETED
ICE602I 0 OPERATION RETURN CODE: 00

UNIQUE Operator Details

 Syntax
UNIQUE FROM(indd) ON(p,m,f) VSAMTYPE(x) UZERO
 ON(VLEN)

 Function

Prints a message in TOOLMSG containing the count of unique values for a specified numeric or character field.

 Examples
UNIQUE FROM(INPUT) ON(20,40,CH)

UNIQUE FROM(DATA) ON(5,3,ZD)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation.

– (p,m,f) gives the position, length and format of a numeric field. A field must not extend beyond position
32752 or the end of the record.

– VLEN is equivalent to ON(1,2,BI) and for variable-length records, represents the record length for each
record.

– A description of each type of field is given below.

 Optional Operands
� VSAMTYPE - the record format for a VSAM input data set (F or V).

� UZERO - causes -0 to be treated as unsigned, that is, as +0.

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

Format Code Length Description

BI 1-256 bytes Unsigned binary

FI 1-256 bytes Signed fixed-point

PD 1-32 bytes Signed packed decimal

ZD 1-32 bytes Signed zoned decimal

CH 1-4000 bytes Character

CSF/FS 1-32 bytes Floating sign

UFF 1-44 bytes Unsigned free form numeric

SFF 1-44 bytes Signed free form numeric

VLEN n/a Record length for VLR (1,2,BI)

 Notes
� The DFSORT DYNALLOC option is used to ensure that work space is available for the sort performed for the

UNIQUE operation.

� Tape work data sets cannot be used with ICETOOL.

Example of TOOLMSG Output for UNIQUE
* UNIQUE - prints a message in TOOLMSG containing the count of
* unique values for a specified numeric or character field.
* Example: print the count of unique values in the
* OUT1 data set ON field.
UNIQUE FROM(OUT1) ON(30,2,PD)

ICE627I 0 DFSORT CALL 0014 FOR SORT FROM OUT1 TO E35 EXIT COMPLETED
ICE628I 0 RECORD COUNT: 000000000000040
ICE610I 0 NUMBER OF UNIQUE VALUES FOR (30,2,PD) : 000000000000011
ICE602I 0 OPERATION RETURN CODE: 00

VERIFY Operator Details

 Syntax
VERIFY FROM(indd) ON(p,m,f) NOSIGN LIMIT(n) VSAMTYPE(x)

 Function

Examines up to 10 specified decimal fields in a data set and prints a message in TOOLMSG identifying each
invalid value found for each field. A decimal value is considered invalid under one of the following circumstances:

� it contains A-F as a digit (example: a PD field of 00AF)

� it contains 0-9 as a sign and the NOSIGN operand is not specified (example: a ZD field of F235).

 Examples
VERIFY FROM(NEW) ON(22,16,PD) ON(7,9,PD)

VERIFY FROM(DATA) ON(28,5,PD) ON(28000,18,ZD) ON(4,7,PD) -
 NOSIGN LIMIT(10)

 Required Operands
� FROM - the ddname of the input data set. You must supply a DD statement for the ddname you specify.

� ON - a field to be used for this operation.

– (p,m,f) gives the position, length and format of a numeric field. A field must not extend beyond position
32752 or the end of the record.

– A description of each type of field is given below.

 Optional Operands
� NOSIGN - the sign of the decimal values is not to be checked for validity.

� LIMIT - a limit for the number of invalid decimal values (overriding the default of 200). If n invalid decimal
values are found, ICETOOL terminates the operation.

� VSAMTYPE - the record format for a VSAM input data set (F or V).

 Symbols
� Symbols for fields can be used instead of p,m,f and p,m in the ON operand.

Format Code Length Description

PD 1-16 bytes Signed packed decimal

ZD 1-18 bytes Signed zoned decimal

 Notes
� Values with invalid digits are also identified for the DISPLAY, OCCUR, RANGE, and STATS operators.

� For each invalid digit found, ICETOOL identifies the relative record number in which the field appears and the
value of the field (in hexadecimal).

� The DISPLAY operator can be used to print a report identifying the relative record number, hexadecimal value
and associated fields for each invalid (and valid) decimal value. See DFSORT Application Programming
Guide, under "DISPLAY Operator", for an example.

Example of TOOLMSG Output for VERIFY
* VERIFY - examines specified decimal fields in a data set and
* prints a message in TOOLMSG identifying each invalid value
* found for each field.
* Example: identify all values in the two IN2 data set
* decimal ON fields that have invalid digits (A-F)
* and/or invalid signs (0-9).
VERIFY FROM(IN2) ON(10,2,ZD) ON(41,6,PD)

ICE618A 0 INVALID (10,2,ZD) VALUE - RECORD: 000000000000003,
 HEX VALUE FAF2
ICE618A 0 INVALID (10,2,ZD) VALUE - RECORD: 000000000000006,
 HEX VALUE F134
ICE618A 0 INVALID (41,6,PD) VALUE - RECORD: 000000000000007,
 HEX VALUE 000000105739
ICE627I 0 DFSORT CALL 0015 FOR COPY FROM IN2 TO E35 EXIT COMPLETED
ICE628I 0 RECORD COUNT: 000000000000008
ICE602I 0 OPERATION RETURN CODE: 12

 Symbol Processing

A DFSORT symbol is a name (preferably something meaningful) that represents a field or a constant. Sets of
symbols, also called mappings, can be used to describe a group of related fields and constants such as the informa-
tion in a particular type of record. Such mappings allow you to refer to fields and constants by their symbols,
freeing you from having to know the position, length and format of a field or the value of a constant you want to
use.

DFSORT's symbol processing feature gives you a powerful, simple and flexible way to create symbol mappings for
your own frequently used data. In addition, you can obtain IBM-created symbol mappings and sample jobs for data
associated with RACF, DFSMSrmm and DCOLLECT. For details, visit the DFSORT home page at URL:

http://www.ibm.com/storage/dfsort

DFSORT symbols can be up to 50 characters, are case-sensitive and can include underscores and hyphens. Thus,
you can create meaningful, descriptive names for your symbols, such as Price_of_Item (or Price-of-Item), making
them easy to remember, use and understand.

You can define and use a symbol for any field or constant in the following ICETOOL operators: COUNT,
DATASORT, DISPLAY, OCCUR, RANGE, SELECT, SPLICE, STATS, SUBSET, UNIQUE and VERIFY. You
can also use symbols in the DFSORT control statements you specify for an ICETOOL run. Two DD statements are
used for symbol processing as follows:

� SYMNAMES: activates symbols processing and defines the symbol data sets to be used.

� SYMNOUT: can be used to define a data set in which the original symbols and the symbol table built by
ICETOOL will be displayed.

Here's a sample ICETOOL job that shows how symbols can be used for ICETOOL operators and DFSORT control
statements. See DFSORT Application Programming Guide for a complete explanation of DFSORT Symbols.

//TOOLSYM JOB ...
//DOIT EXEC PGM=ICETOOL
//TOOLMSG DD SYSOUT=*
//DFSMSG DD SYSOUT=*
//SYMNAMES DD DSN=MY.CUSTOMER.SYMBOLS,DISP=SHR
//SYMNOUT DD SYSOUT=*
//IN DD DSN=MY.CUSTOMER.INPUT,DISP=SHR
//OUT DD DSN=&O,UNIT=SYSDA,SPACE=(CYL,(5,5),RLSE),
// DISP=(,PASS)
//LIST1 DD SYSOUT=*
//TOOLIN DD *
RANGE FROM(IN) ON(Customer_Balance) LOWER(Stop_Check)
SORT FROM(IN) TO(OUT) USING(CTL1)
DISPLAY FROM(OUT) LIST(LIST1) BLANK WIDTH(133) -
TITLE(Title) DATE(4MD/) PAGE -
HEADER(Head1) ON(Customer_Name) -
HEADER(Head2) ON(Customer_Balance,C1) -

 HEADER(Head3) ON(Customer_Flags,HEX)
/*
//CTL1CNTL DD *
 SORT FIELDS=(Customer_Balance,D,Customer_Name,A)
 INCLUDE COND=((Dept_Code,EQ,Research,OR,
 Dept_Code,EQ,Marketing),
 AND,Customer_Balance,GT,Gift)

Calling ICETOOL from a Program

ICETOOL can be called from an assembler program using LINK, ATTACH, or XCTL, and standard linkage con-
ventions.

When all operators have been processed, ICETOOL returns to the calling program with register 15 (R15) set to the
highest operation return code encountered.

Two different interfaces are available:

 � TOOLIN Interface:

You supply ICETOOL statements in the TOOLIN data set. ICETOOL prints messages in the TOOLMSG data
set, but does not return information directly to your program.

� Parameter List Interface

You supply ICETOOL statements in a parameter list. ICETOOL prints messages in the TOOLMSG data set
and also returns information in the parameter list for use by your program.

For complete details of ICETOOL's calling program interface, see DFSORT Application Programming Guide.

Appendix A. Edit Masks for DISPLAY and OCCUR
Formatting

The table below describes the available masks and shows how the values 12345678 and -1234567 would be printed
for each mask. In the pattern:

� d is used to represent a decimal digit (0-9)

� w is used to represent a leading sign that will be blank for a positive value or - for a negative value

� x is used to represent a trailing sign that will be blank for a positive value or - for a negative value

� y is used to represent a leading sign that will be blank for a positive value or ( for a negative value

� z is used to represent a trailing sign that will be blank for a positive value or ) for a negative value

Mask Pattern 12345678 -1234567

A0 wddddddddddddddddddddddddddddddd 12345678 -1234567

A1 wd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd 12,345,678 -1,234,567

A2 wd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd 12.345.678 -1.234.567

A3 wd ddd ddd ddd ddd ddd ddd ddd ddd ddd ddd 12 345 678 -1 234 567

A4 wd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd 12'345'678 -1'234'567

A5 d ddd ddd ddd ddd ddd ddd ddd ddd ddd dddx 12 345 678 1 234 567-

B1 wddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd.d 1,234,567.8 -123,456.7

B2 wddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd,d 1.234.567,8 -123.456,7

B3 wddd ddd ddd ddd ddd ddd ddd ddd ddd ddd,d 1 234 567,8 -123 456,7

B4 wddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd.d 1'234'567.8 -123'456.7

B5 wddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd,d 1'234'567,8 -123'456,7

B6 ddd ddd ddd ddd ddd ddd ddd ddd ddd ddd,dx 1 234 567,8 123 456,7-

C1 wdd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd.dd 123,456.78 -12,345.67

C2 wdd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd,dd 123.456,78 -12.345,67

C3 wdd ddd ddd ddd ddd ddd ddd ddd ddd ddd,dd 123 456,78 -12 345,67

C4 wdd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd.dd 123'456.78 -12'345.67

C5 wdd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd,dd 123'456,78 -12'345,67

C6 dd ddd ddd ddd ddd d ddd ddd ddd ddd,ddx 123 456,78 12 345,67-

D1 wd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd.ddd 12,345.678 -1,234.567

D2 wd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd,ddd 12.345,678 -1.234,567

D3 wd ddd ddd ddd ddd ddd ddd ddd ddd ddd,ddd 12 345,678 -1 234,567

D4 wd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd.ddd 12'345.678 -1'234.567

D5 wd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd,ddd 12'345,678 -1'234,567

D6 d ddd ddd ddd ddd ddd ddd ddd ddd ddd,dddx 12 345,678 1 234,567-

Mask Pattern 12345678 -1234567

E1 yd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,dddz 12,345,678 (1,234,567)

E2 yd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.dddz 12.345.678 (1.234.567)

E3 yd ddd ddd ddd ddd ddd ddd ddd ddd ddd dddz 12 345 678 (1 234 567)

E4 yd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'dddz 12'345'678 (1'234'567)

F1 ydd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd.ddz 123,456.78 (12,345.67)

F2 ydd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd,ddz 123.456,78 (12.345,67)

F3 ydd ddd ddd ddd ddd ddd ddd ddd ddd ddd,ddz 123 456,78 (12 345,67)

F4 ydd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd.ddz 123'456.78 (12'345.67)

F5 ydd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd,ddz 123'456,78 (12'345,67)

G1 wddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd,ddd.dddd 1,234.6678 -123.4567

G2 wddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd.ddd,dddd 1.234,5678 -123,4567

G3 wddd ddd ddd ddd ddd ddd ddd ddd ddd,dddd 1 234,5678 -123,4567

G4 wddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd.dddd 1'234.5678 -123.4567

G5 wddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd'ddd,dddd 1'234,5678 -123,4567

G6 ddd ddd ddd ddd ddd ddd ddd ddd ddd,ddddx 1 234,5678 123,4567-

